
บทที่   Ä
กรณีศึกษาระบบงานทะเบียน

จดุประสงคเชิงพฤติกรรม
1. เขาใจระบบงานของส ํานักทะเบียน ในมุมมองของ Programmer มากขึ้น
2. เขาใจการสรางระบบงานคอมพิวเตอร โดยพิจารณาจากความตองการของผูใช
3. มปีระสบการณ และมองเห็นแนวการออกแบบเมนูทั้งระบบงาน
4. เขาใจแนวการเขียน และการแปลโปรแกรม โดยใชวิธีการอยางงาย
5. เขาใจการท ํารายงาน และการปรับปรุงขอมูลในรูปแบบตาง ๆ
6. เขาใจการเขียนเมน ูและเรียกใชงานอยางงาย

หวัขอในบทเรียน
5.1  ศึกษาความตองการ
5.2  โครงสรางขอมูล
5.3  วิเคราะหระบบงานเบื้องตน
5.4  โปรแกรมในระบบงานทะเบียน

5.4.1  โปรแกรมตาง ๆ ใน MAIN.PRG     : สารบัญของโปรแกรม
5.4.2  โปรแกรมตาง ๆ ใน SUB01.PRG  : รวมโปรแกรมเตรียมขอมูล และการลงทะเบียน
5.4.3  โปรแกรมตาง ๆ ใน SUB02.PRG  : รวมโปรแกรมรายงานหลังลงทะเบียน
5.4.4  โปรแกรมตาง ๆ ใน SUB03.PRG  : รวมรายงานตามคํ ารองของผูใช
5.4.5  โปรแกรมตาง ๆ ใน SUB04.PRG  : รวมรายงานประจํ าภาคเรียน
5.4.6  โปรแกรมตาง ๆ ใน SUB05.PRG  : รวมโปรแกรมปรับปรุงขอมูล

ตวัอยางโปรแกรมเกี่ยวกับระบบงานทะเบียน   38  ตัวอยาง

หมายเหตุ
ตัวอยางในบทนี้ประยุกตจากประสบการณที่ผูเขียนทํ างานเขียนโปรแกรมใหสํ านักทะเบียนโยนก ซึ่งไดมีการ

ประสานงานกบัคณุลัดดาวรรณ  ปญญา ซึ่งเปนผูปฏิบัติงาน แลวเรียบเรียงใหม ใหเหมาะสมกับเนื้อหาวิชา และเวลาที่
จะใชสอน ใหดเูหมือนเปนระบบที่อาจน ําไปใชได แตในความเปนจริงแลว เปนเพียงกรณีศึกษา เพื่อชี้ใหเห็นถึงจุดที่ตอง
ปรับปรุงเพิ่มเติมอีกมาก .. ประกอบการบรรยายในชั้นเรียน และโปรแกรมทั้งหมดในหนังสือ นักศึกษาสามารถ 
Download ไปศกึษา ประกอบการเรียนภาคปฏิบัติ

บทที่ 5   กรณีศึกษาระบบงานทะเบียน
เนือ่งจากระบบทะเบียนทางการศึกษา ซึ่งเก็บประวัติการเรียนของนักเรียน เปนระบบที่น ํามาใชเก็บประวัติการ

ศกึษาในทกุสถาบนั โดยแตละสถาบันมีรายละเอียดบางประการที่ไมเหมือนกัน  แตแกนแทของงานทะเบียนแลวใกล
เคยีงกนั คอืงานบริการขอมูลการศึกษาส ําหรับนักศึกษา อาจารยและผูบริหารของสถานศึกษานั้น ๆ

ดวยเหตทุีผู่สนใจศึกษาการเขียนโปรแกรมทุกคน  ตองผานการศึกษามาอยางนอยหนึ่งสถาบัน และไดเขาไป
เกีย่วของกบังานทะเบียน หรือมีรายชื่อของตนในสถาบันนั้น และไดรับรายงานเชน ใบแจงผลการเรียน  ท ําใหทุกคนได
คุนเคย และมองภาพรวมของงานทะเบียน วาระบบนี้ควรมีงานอะไรที่ตองทํ าบาง

ดงันัน้ผูเขยีนจงึของน ําระบบทะเบียนมาเปนกรณีศึกษา  เพื่อใหผูอานไดเขาใจในเนื้อหาของระบบงาน  เพื่อ
น ําไปสูหลกัการพัฒนาโปรแกรมในระบบอื่น ๆ ตอไป ซึ่งผูเขียนขอเริ่มตั้งแตการศึกษาความตองการของผูใช แลวน ํามา
วเิคราะหหาความตองการ  ที่จะน ํามาเปนแนวทางในการพัฒนาเปนระบบยอยตาง ๆ จนถึงการเขียนโปรแกรมใหได
ตามความตองการ  12

&  5.1 ศึกษาความตองการ
แผนกทะเบยีนทีจ่ะศึกษาตอไปนี ้  ไมไดเนนที่จะศึกษาจนครอบคลุมทุกอยางที่แผนกทะเบียนทํ า แตจะศึกษา

เฉพาะสวนงานที่เกี่ยวกับการลงทะเบียน จัดเก็บเกรด และรายงานเกรดใหนักศึกษาทราบ

5.1.1 บุคคลที่เกี่ยวของกับงาน
ลํ าดับ ผูเกี่ยวของ ความสัมพันธตอขอมูล
1 นกัศึกษา  มาลงทะเบียนเรียน และรอรับผลการเรียนเมื่อสิ้นภาคการศึกษา
2 อาจารยผูสอน เปนผูระบุวิชาที่ตองการสอน รอรับรายชื่อ และสงเกรดทายภาค
3 ผูบริหาร เปนผูใหนโยบาย และรับรายงานสรุปผลตาง ๆ
4 แผนกทะเบียน ท ําหนาที่ติดตอกับฝายตาง ๆ เพื่อน ําขอมูลเขาระบบ และพิมพรายงาน

5.1.2 ระบบงานที่เปนอยู และสิ่งที่ผูใชตองการ
ความตองการของผูใชมีมากมาย โดยเฉพาะการพูดคุยกับผูใชที่ไมรูจักคอมพิวเตอร จํ าเปนตองตีความหมาย

ระบบงานทีเ่ปนอยู และเรียบเรียงใหมเปนขอ ๆ  แลวใหผูใชตรวจสอบอีกครั้งหนึ่งวาสิ่งที่ผูใชตองการ กับทีมพัฒนา
ระบบเขาใจตรงกันหรือไม

บอยครัง้ทีก่ารพูดคุยระหวางทีมพัฒนาระบบ กับกลุมผูใช   ถึงปญหาในระบบงานเดิม  ทํ าใหตองปรับปรุงการ
ท ํางานแบบเดมิในหลาย ๆ จุด เพราะในการน ําระบบคอมพิวเตอรเขามาใชนั้น  การท ํางานตองมีขั้นตอนที่ชัดเจน และ
ตรวจสอบความถูกตองได   การทะเลาะและไมเขาใจกันจึงเกิดขึ้นจนเปนเรื่องปกติในระหวางการพัฒนาระบบงานใหม
แตทัง้สองฝายควรทํ าความเขาใจซึ่งกันและกัน  เพื่อความสมบูรณของระบบใหมที่จะนํ ามาใช หลังจากการพัฒนาเสร็จ
สิ้นลง

12 ระบบทะเบยีนนีอ้างอิงจากประสบการณของผูเขียน ที่ท ํางานในสํ านักทะเบียนวิทยาลัยโยนก แลวประยุกต
มาเปนรูปแบบของ Clipper ซึง่เดมิพัฒนาดวย COBOL แลวเปลี่ยนมาเปน Microsoft Access

บางครัง้จ ําเปนตองเปลี่ยนแปลงระบบงานที่เปนอยูใหมีความเปนระเบียบ   เพื่อจะไดจัดเก็บขอมูลและ
ประมวลผลในคอมพวิเตอรไดอยางถูกตอง  รายการตอไปนี้เปนความตองการของผูใชที่ไดปรับปรุงใหเขาใจไดงาย เปน
ขอ ๆ โดยศึกษามาจากระบบงานที่มีอยู
ความตองการของผูใช
1. อาจารยผูสอนใหขอมูลการเปดวิชา กอนเปดภาคการศึกษาใหม 1 เดือน
2. น ําขอมูลนักศึกษาใหมเขาฐานขอมูล กอนเปดเรียน
3. การลงทะเบียน มีการตรวจสอบจํ านวนไมใหเกินที่กํ าหนด
4. การขอแกไข บอกเพิ่ม-ลดวิชา 1 สัปดาหหลังวันลงทะเบียน
5. พมิพขอมูลการลงทะเบียนติดประกาศ ใหนักศึกษาตรวจสอบอีกครั้งหนึ่ง
6. พมิพรายชื่อนักเรียนใหอาจารยผูสอน
7. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
8. การขอเพิกถอนวิชา ของนักเรียน
9. พมิพรายชื่อแตละวิชาติดหนาหองสอบ และเลขที่โตะของนักศึกษา
10. การสงเกรดของอาจารย เมื่อสิ้นภาคเรียน
11. พมิพใบรายงานผลการเรียนประจํ าภาค
12. ปรบัปรงุขอมูลอาจารย เมื่อรับอาจารยใหม ลาออก หรือแตงงาน
13. พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
14. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช
15. แสดงตารางเรียนของนักศึกษา เมื่อผูปกครองมาขอพบ
16. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
17. พมิพรายชื่อแยกตามสถานภาพ
18. พมิพรายชื่อแยกตามสาขา
19. พมิพใบรายงานผลการเรียนตลอดหลักสูตร เมื่อนักศึกษารองขอ
20. การปรบัปรุงและพิมพขอมูลจากแฟมมาตรวจสอบ

&  5.2 โครงสรางขอมูล
เมือ่ทราบถงึสิ่งที่ตองการ  ทีมพัฒนาระบบจะออกแบบแฟมขอมูลส ําหรับเก็บขอมูลหลาย ๆ แฟมซึ่งรวมกัน

แลวเรยีกวาฐานขอมูล ซึ่งจะเก็บขอมูลเพื่อรองรับกับสิ่งที่ผูใชตองการ  จึงตองมีการลองผิดลองถูก หากกลุมแฟมที่จัด
ขึน้ยงัไมสามารถเก็บขอมูลเพื่อสนองสิ่งที่ผูใชตองการได จะตองแกไขใหมจนกวาจะไดกลุมแฟมที่รองรับงานของผูใชได
หมด

แฟมที่เกี่ยวของกับระบบงานทะเบียน
1. แฟมนักศึกษา   IDSTD.DBF
   1.1 รหัส             *IDSTD         9(7) ตวัอยางขอมูล
      (2หลกัแรก:ปที่เขา 2หลักตอมา:คณะที่เรียน 3หลักสุดทาย:ล ําดับที่) 3901001 นาย สมคิด  สมควร  11 A
  1.2 ชื่อ              INAME         X(20) 3901002 น.ส.ถนัด   ประดิษฐ 14  A
  1.3 สกุล             ISURN         X(20) 3901003 นาย สมบัต ิภูเบศร   12  D
  1.4 สาขา              IMAJOR        99
      (11:บญัช ี12:การจัดการ 13:การตลาด 14:คอมพิวเตอรธุรกิจ)
  1.5 สถานภาพ           ISTAT         X
      (A:ก ําลังศึกษาอยู   D:ลาออก)
2.  แฟมวิชา     SUBJECT.DBF
  2.1 ชือ่วิชา         *SUBJ          X(8) ตวัอยางขอมูล
     (4หลกัแรก:ชื่อหมวด 3หลักหลัง:ชื่อล ําดับของวิชาในหมวด) SOCI 101 3930005 4201 09.00-11.50 28 30
  2.2 รหัสผูสอน        TEACHID       9(7) BCOM101 3914007 1203 08.00-10.50 6 6
(2หลกัแรก:ปที่เขา  ตอมา:หนวยงานที่ประจํ า 3หลักสุดทาย:ล ําดับ) ENGL202  3930002 1211 14.00-16.50 25 25
  2.3 เลขที่หอง        ROOM          X(4)
      (เบอรหองที่ปรากฏหนาหองตามตึกตาง ๆ)
  2.4 เวลาเรียน         STIME         X(11)
      (เชน 09.00-11.50 เริ่มเรียน จนถึงเลิกเรียน)
  2.5 จ ํานวนลงทะเบียน    SREGIST       999
  2.6 จ ํานวนรับได       SLIMIT        999
3.  แฟมอาจารยผูสอน   TEACHER.DBF
  3.1 รหัสผูสอน       *TEACHID       9(7) ตวัอยางขอมูล
 (2หลกัแรก:ปที่เขา  ตอมา:หนวยงาน  3หลักสุดทาย:ล ําดับที่) 3910001 นาย อติชาติ     ปรีชา         40  M
  3.2 ชื่อผูสอน        TNAME         X(20) 3914007 น.ส.ศริิลักษณ  เกศรินทร     45  F
  3.3 สกุลผูสอน        TSURN         X(20) 3930005 นาย จักรพันธ   สุวรรณ       34  M
  3.4 อายุ             TAGE          999
  3.5 เพศ               TSEX          X

(M:ผูชาย F:ผูหญิง)

4.  แฟมลงทะเบียน   REGIST.DBF
4.1 ปการศึกษา       *RYEAR         9999  ตวัอยางขอมูล
4.2 ภาคเรียน         *RSEM          X 2002  1  3901001   BCOM 101   4.0
4.3 รหัสนักศึกษา     *IDSTD         9(7) 2002  1  3901001   SOCI 101     2.5
4.4 ชือ่วิชา         *SUBJ          X(8) 2002  1  3901001   ENGL 101    2.0
4.5 เกรดที่ไดรับ     RGRADE        9V9 2002  1  3901002   BCOM 101   1.5
(0:F  1:D  1.5:D+  2:C  2.5:C+  3:B  3.5:B+  4:A) 2002  1  3901002   ENGL 201    1.0

 2002  1  3901003   BCOM 301   3.5
2002  1  3901003   MATH 303   3.0
2002  2  3901001   GOVT 101    3.0
2002  2  3901001   ENGL 102    3.5
2002  2  3901003   ARTS 201    1.0
2002  2  3901003   COMM 401  1.5
2002  2  3901003   BCOM 305   2.0

 5.  แฟมหอง    ROOM.DBF
5.1 เลขที่หอง       *ROOM          X(4) ตวัอยางขอมูล
5.2 ล ําดับโตะ         RCSEQ         999  1201  1   A1
5.3 เลขที่โตะ        RCHAIR        XX  1201  2   A2
(แตละหองเลขโตะเรียงไมเหมือนกัน เชน A1,A2,A3,...)  1201  3   A3

  1201  4   B1
1201  5   B2
1202  1   A1
1202  2   A2
1202  3   B1

&  5.3  วิเคราะหระบบงานเบื้องตน
การวเิคราะหระบบงานเบื้องตน ในครั้งนี้จะสนใจในเรื่องของสิ่งที่ผูใชตองการ แลวน ํามาแยกเปนระบบเล็ก

เพือ่ทีจ่ะจบักลุมใหกระชับ และน ําไปออกแบบเมน ู แลวน ําไปเปนแนวทางในการจัดทํ าโปรแกรม  แตหากนํ าไปใชงาน
จรงิจะตองมกีารศึกษาความเปนไปได  ผังงาน ความสัมพันธระหวางขอมูล และตนทุนใหชัดเจนกวานี้

5.3.1 จัดระบบยอยจากความตองการของผูใช
ระบบเตรียมงานและงานลงทะเบียน
   1. อาจารยผูสอนใหขอมูลการเปดวิชา กอนเปดภาคการศึกษาใหม 1 เดือน
   2. ปรบัปรงุขอมูลนักศึกษาเดิม และน ําขอมูลนักศึกษาใหมเขาฐานขอมูล กอนเปดเรียน
   3. การลงทะเบียน มีการตรวจสอบจํ านวนไมใหเกินที่กํ าหนด
   4. การขอแกไข บอกเพิ่ม-ลดวิชา 1 สัปดาหหลังวันลงทะเบียน
ระบบงานหลังลงทะเบียน
   1. พมิพขอมูลการลงทะเบียนติดประกาศ ใหนักศึกษาตรวจสอบอีกครั้งหนึ่ง
   2. พมิพรายชื่อนักเรียนใหอาจารยผูสอน
   3. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
   4. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช
   5. พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
   6. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
ระบบงานเมื่อถูกรองขอ
   1. การขอเพิกถอนวิชา ของนักเรียน
   2. แสดงตารางเรียนของนักศึกษา เมื่อผูปกครองมาขอพบ
   3. พมิพใบรายงานผลการเรียนตลอดหลักสูตร เมื่อนักศึกษารองขอ
   4. ปรบัปรงุขอมูลอาจารย เมื่อรับอาจารยใหม ลาออก หรือแตงงาน
ระบบงานประจํ าภาคเรียน
   1. พมิพรายชื่อแตละวิชาติดหนาหองสอบ และเลขที่โตะของนักศึกษา
   2. พมิพรายชื่อแยกตามสถานภาพ
   3. พมิพรายชื่อแยกตามสาขา
   4. การสงเกรดของอาจารย เมื่อสิ้นภาคเรียน
   5. พมิพใบรายงานผลการเรียนประจ ําภาค
ระบบปรับปรุงและพิมพขอมูลจากแฟม
   1. แฟมนักศึกษา 2. แฟมวิชา
   3. แฟมผูสอน 4. แฟมลงทะเบียน
   5. แฟมหอง

5.3.2 ออกแบบเมนูตามระบบยอย
เมนูหลัก
1. ระบบเตรียมงานและงานลงทะเบียน
2. ระบบงานหลังลงทะเบียน
3. ระบบงานเมื่อถูกรองขอ
4. ระบบงานประจํ าภาคเรียน
5. ระบบปรับปรุงและพิมพขอมูลจากแฟม
6. ขอมูลระบบ และทีมพัฒนาโปรแกรม
7. ออกไประบบปฏิบัติการชั่วคราว
8. เลกิการท ํางาน
1. เมนรูะบบเตรียมงานและงานลงทะเบียน
   1. รบัขอมูลการเปดวิชาใหม
   2. รบัขอมูลขอมูลนักศึกษาใหม
   3. รบัลงทะเบียนจากนักศึกษา
   4. รบับอกเพิ่ม-ลดวิชา
   5. กลับเมนูหลัก
2. เมนูระบบงานหลังลงทะเบียน
   1. พมิพขอมูลการลงทะเบียนติดประกาศ
   2. พมิพรายชื่อนักเรียนใหอาจารยผูสอน
   3. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
   4. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช
   5. พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทาน
   6. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
   7. กลับเมนูหลัก
3. เมนรูะบบงานเมื่อถูกรองขอ
   1. การขอเพิกถอนวิชา
   2. ตรวจสอบตารางเรียนของนักศึกษาเฉพาะคน
   3. พมิพรายงานผลการเรียนตลอดหลักสูตร
   4. ปรับปรุงขอมูลอาจารย
   5. กลับเมนูหลัก

4. เมนรูะบบงานประจ ําภาคเรียน
   1. พมิพรายชื่อติดหนาหองสอบ
   2. พมิพรายชื่อแยกตามสถานภาพ
   3. พมิพรายชื่อแยกตามสาขา
   4. ใสเกรด
   5. พมิพรายงานผลการเรียนประจ ําภาค
   6. กลับเมนูหลัก
5. เมนูระบบปรับปรุงและพิมพขอมูลจากแฟม
   1. แฟมนักศึกษา
   2. แฟมวิชา
   3. แฟมผูสอน
   4. แฟมลงทะเบียน
   5. แฟมหอง
   6. กลับเมนูหลัก

5.3.3 หลักการตั้งชื่อโปรแกรม
การตัง้ชือ่โปรแกรมมีหลายวิธ ีและแตละวิธีมีจุดเดน และจุดดอยที่แตกตางกัน ขึ้นกับลักษณะหนวยงาน

ลกัษณะบุคลากร และขอตกลงของทีมพัฒนาโปรแกรม
หลกัในการตั้งชื่อมีหลายวิธี
1. ตัง้ชือ่ตามลักษณะงานที่ใช
2. ตัง้ชือ่ตามหนวยงานที่เกี่ยวของ และขอมูลที่เกี่ยวของ
3. ตัง้ชือ่ตามลํ าดับในเมนู
4. ตัง้ชื่อตามผูรับผิดชอบโปรแกรม
5. ตัง้ชือ่ตามวันที่ที่เริ่มเขียนโปรแกรม
6. ตัง้ชือ่ตามลํ าดับของโปรแกรม
7. อื่น ๆ อีกมาก

   ส ําหรบัระบบงานทะเบียน ขอตั้งชื่อตามลํ าดับของตัวเลือกในเมนู เพื่อสื่อใหเขาใจไดโดยงาย   ส ําหรับชื่อของ
โปรแกรมกํ าหนดไว 5 หลัก โดยแตละหลักมีความหมายดังนี้

หลกัที ่1 ใหเปนตัว R ซึ่งยอมาจาก REGISTRA
หลักที ่2 และ 3 เปนล ําดับในเมนูหลัก
หลกัที ่4 และ 5 เปนล ําดับในเมนูยอย

โปรแกรมตาง ๆ ในระบบทะเบียน
MAIN.PRG : เมนหูลกัส ําหรับระบบงานทะเบียน เพื่อท ําตัวเลือกหลักทั้งหมด ประกอบดวย 21 โปรแกรมยอย
MAINALL : แสดงเมนูหลักบนจอภาพ
SUB1 : แสดงเมนูยอย ระบบเตรียมงาน และงานลงทะเบียน
SUB2 : แสดงเมนูยอย ระบบงานหลังลงทะเบียน
SUB3 : แสดงเมนูยอย ระบบงานเมื่อถูกรองขอ
SUB4 : แสดงเมนูยอย ระบบงานประจ ําภาคเรียน
SUB5 : แสดงเมนูยอย ระบบปรับปรุง และพิมพขอมูลจากแฟม
WORKSYSTEM : แสดงขอมูลของระบบทางจอภาพ
ASKPRT : สอบถามการปลายทางของผลลัพธ
ASKUPD : ถามยํ ้าการปรับปรุงขอมูล
SORTREG1 : จดัเรยีงแฟม REGIST ตามฟลด RYEAR,RSEM,IDSTD,SUBJ
SORTREG2 : จดัเรียงแฟม REGIST ตามฟลด SUBJ,RYEAR,RSEM,IDSTD
SORTSUB1 : จดัเรียงแฟม SUBJECT ตามฟลด ROOM,SUBJ
SORTSUB2 : จดัเรียงแฟม SUBJECT ตามฟลด SUBJ
SORTID1 : จดัเรียงแฟม IDSTD ตาม IDSTD
SORTID2 : จดัเรยีงแฟม IDSTD ตาม ISTAT,IDSTD
SORTID3 : จดัเรียงแฟม IDSTD ตาม IMAJOR,INAME,ISURN
GRADECHR : แปลงเกรดจากตัวเลขเปนตัวอักษร
GRADENUM : แปลงเกรดจากตัวอักษรเปนตัวเลข
HELPFORM : แสดงขอความชวยเหลืออยางงาย ๆ เมื่อกดปุมฟงกชัน F1
HELPREADVAR : แสดงคาที่เปนไปไดของฟลดที่ก ําลังรอรับคา
QUITPROC : สัง่เลกิท ํางานของโปรแกรมทั้งหมด
SUB01.PRG : ระบบเตรียมงานและลงทะเบียน  ประกอบดวย 4 โปรแกรมยอย
R0101 : ปรบัปรงุแฟมวิชา ดวยขอมูลการเปดวิชา จากอาจารยผูสอน
R0102 : เพิม่รายชือ่นักศึกษาเขาใหมในแฟมนักศึกษา
R0103 : เพิม่ขอมูลการลงทะเบียนในแฟมลงทะเบียน
R0104 : ปรบัปรุงขอมูลการลงทะเบียนในแฟมลงทะเบียน
SUB02.PRG : ระบบงานหลังลงทะเบียน  ประกอบดวย 6 โปรแกรมยอย
R0201 : พมิพขอมูลการลงทะเบียนติดประกาศ
R0202 : พมิพรายชื่อนักเรียนใหอาจารยผูสอน
R0203 : พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร
R0204 : พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช

R0205 : พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
R0206 : พมิพสรุปจ ํานวนนักศึกษาแตละวิชา
SUB03.PRG : ระบบงานเมื่อถูกรองขอ  ประกอบดวย 4 โปรแกรมยอย
R0301 : ปรบัปรงุเกรดนักศึกษาที่ขอเพิกถอนวิชา
R0302 : พมิพตารางเรียนของนักศึกษาแตละวิชา
R0303 : พมิพใบรายงานผลการเรียนตลอดหลักสูตร
R0304 : ปรับปรุงแฟมอาจารย
SUB04.PRG : ระบบงานประจํ าภาคเรียน  ประกอบดวย 5 โปรแกรมยอย
R0401 : พมิพรายชื่อนักศึกษาแตละวิชาติดหนาหองสอบ
R0402 : พมิพรายชื่อแยกตามสถานภาพ
R0403 : พมิพรายชื่อแยกตามสาขา และจัดเรียงตามชื่อ
R0404 : ปรบัปรงุเกรดนักศึกษา เมื่ออาจารยสงเกรด
R0405 : พมิพใบรายงานผลการเรียนประจํ าภาค
SUB05.PRG : ระบบปรับปรุงและพิมพขอมูลจากแฟม  ประกอบดวย 5 โปรแกรมยอย
R0501 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมนักศึกษา
R0502 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมวิชา
R0503 : เพิ่ม ลบ แกไขหรือพิมพขอมูลจาก แฟมผูสอน
R0504 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมลงทะเบียน
R0505 : เพิม่ ลบ แกไขหรือพิมพขอมูลจาก แฟมหอง

5.3.4 วิธีการแปลโปรแกรม
การแปลโปรแกรม ในกรณีศึกษานี้ ใชคํ าสั่ง RMAKE ในการแปลโปรแกรม  โดยระบุชื่อโปรแกรมที่จะแปลไว

ในแฟม MAIN.RMK เพราะการแปลแบบนี้จะท ําใหการพัฒนาเร็วขึ้น  หากมีการเรียกโปรแกรมยอยจากภายนอกหลาย
ๆ โปรแกรม  เมื่อแกโปรแกรมยอยบางโปรแกรมจะตองแปลโปรแกรมทั้งหมดใหมดวยการแปลวิธีเดิม แตการแปลดวย
RMAKE จะเลอืกเฉพาะโปรแกรมยอยที่ถูกแกไขมารับการแปลใหมเทานั้น

จดุเดนของการใช  RMAKE.EXE  ชวยคือ การแปลโปรแกรมแตละครั้งจะเลือกเฉพาะโปรแกรมยอยที่ถูกปรับ
ปรงุมาแปลใหม  สวนที่ไมไดแกไขจะใชของแฟมเดิม  ท ําใหเร็วขึ้นอยางมากในการพัฒนาโปรแกรมที่มีการเรียกใช
โปรแกรมยอยจากภายนอกหลาย ๆ  โปรแกรม

การสรางแฟม MAIN.RMK
//  MAIN.RMK
//  ========
.PRG.OBJ:
      CLIPPER $< /M /N
.OBJ.EXE:
      SET RTLINKCMD=/POSI

       RTLINK $**, $@ ;
MAIN.OBJ:     MAIN.PRG
SUB01.OBJ:    SUB01.PRG
SUB02.OBJ:    SUB02.PRG
SUB03.OBJ:    SUB03.PRG
SUB04.OBJ:    SUB04.PRG
SUB05.OBJ:    SUB05.PRG
MAIN.EXE: MAIN.OBJ  SUB01.OBJ SUB02.OBJ SUB03.OBJ SUB04.OBJ SUB05.OBJ

การแปลโปรแกรมจะน ํา MAIN.RMK มารับการแปลดวย RMAKE โดยเขียนที่ DOS PROMPT ดังนี้
C:\CLIPPER5\RMAKE   MAIN      แตยงัคงใชแฟม CLIPPER.EXE และ RTLINK.EXE   เหมือนการแปลดวย
C:\CLIPPER5\CL MAIN

สิง่ทีแ่ตกตางของ RMAKE และ CL คือ RMAKE.EXE จะชวยเลือกโปรแกรมยอยที่มีการแกไขเทานั้นมา
ผานการแปลดวยโปรแกรมชุดเดิม แตไมไดน ําโปรแกรมยอยทั้งหมดมาแปลใหม

การเขยีนโปรแกรมให RMAKE แปล ตองเขียนทุกโปรแกรมใหอยูใน PROCEDURE หรือ FUNCTION  ถึงเปน
โปรแกรมหลักก็ตองมีโปรแกรมแรกอยูใน PROCEDURE แรกของโปรแกรม

&  5.4 โปรแกรมในระบบทะเบียน
โปรแกรมของระบบนี ้ไดจัดทํ าเปนโปรแกรมยอยไวภายนอก โดยมีโปรแกรม MAIN.PRG คอยเรียกโปรแกรม

เลก็ ๆ ที่อยูในโปรแกรมยอยอีกครั้งหนึ่ง  ส ําหรับโปรแกรมยอยภายนอกนั้น แตละโปรแกรมยอยคือระบบยอยระบบหนึ่ง
เพราะท ําใหงายตอการปรับปรุงแกไขและคนหา หากประมวลผลตามตัวเลือกแลวผิดพลาด

โปรแกรมยอยจากภายนอกมี 5 โปรแกรมคือ SUB01.PRG จนถึง SUB05.PRG  รวมโปรแกรมทั้งหมดใน
ระบบนี้เปน 6  โปรแกรม  โดยใหโปรแกรมยอยทั้ง 5 นี้ควบคุมระบบยอย 5 ระบบ

1. ระบบเตรียมงาน และงานลงทะเบียน
2. ระบบงานหลังลงทะเบียน
3. ระบบงานเมื่อถูกรองขอ
4. ระบบงานประจํ าภาคเรียน
5. ระบบปรับปรุงและพิมพขอมูลจากแฟม

5.4.1 โปรแกรมตาง ๆ ใน MAIN.PRG
โปรแกรมใน MAIN.PRG นี้มีโปรแกรมยอยทั้งหมด 21 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับเมนู และ

โปรแกรมสนับสนุน เชน โปรแกรมสอบถามปลายทางของผลลัพธ  โปรแกรมถามยํ้ าการปรับปรุงขอมูล โปรแกรมจัด
เรยีงขอมูล โปรแกรมแปลงเกรด โปรแกรมสั่งเลิกการท ํางาน เปนตน

โปรแกรมเมนูในโปรแกรม MAIN.PRG นี้มีหลักการงาย ๆ โดยสามารถเลือกตัวเลือกดวยคํ าสั่ง PROMPT และ
สงการท ํางานไปเมนูยอย ในขณะที่เมนูยอยแตละเมนูมีระบบการทํ างานที่สมบูรณในตนเอง จึงทํ าใหเมนูแบบนี้ทํ า
ความเขาใจไดงาย และน ําไปใชในทางปฎิบัติไดทันที

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม MAIN.PRG ทั้งหมด 21 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.1 ถงึ ตวัอยางที่ 5.14  (บางโปรแกรมสั้นและมีการท ํางานคลายกัน จึงนํ ามารวมในตัวอยางเดียวกัน)
: ตัวอยางที่ 5.1
&   ผลลัพธ
// โปรแกรมหลัก ของระบบงานทะเบียน เพื่อจัดทํ าเมนูหลัก แลวเลือกตัวเลือก
// เพือ่เขาไปสูเมนูยอย แตมีการระบุโปรแกรมยอยใหกับปุมฟงกชัน โดยระบุให
// ปุมฟงกชัน F1 เรียกโปรแกรมยอย HELPFORM เพื่อแสดงขอมูลผูเขียน
// ปุมฟงกชัน F2 เรียกโปรแกรมยอย HELPREADVAR เพื่อชวยผูใชปอนคาไดงาย
// ปุมฟงกชัน F10 เรียกโปรแกรมยอย QUITPROC เพื่อเลิกการทํ างาน
PROCEDURE MAINALL
   SET KEY 28 TO HELPFORM  // ควบคมุปุม F1 เพื่อแสดงขอมูลเกี่ยวกับผูเขียนโปรแกรม
   SET KEY -1 TO HELPREADVAR // ควบคมุปุม F2 เพื่อเลือกขอมูลไดอยางงาย ๆ
   SET KEY -9 TO QUITPROC    // ควบคมุปุม F10 เพื่อถามการเลิกทํ างาน
   SET WRAP ON
   SET DELIMITERS TO '[ ]'

   SET DELIMITERS ON
   SET EPOCH TO 1960
   DO WHILE .T.
      CLS ; OPT = 1
      @ 8,5 SAY "เมนูหลัก ระบบงานทะเบียน"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. ระบบเตรียมงานและงานลงทะเบียน"
      @ 11,5 PROMPT "2. ระบบงานหลังลงทะเบียน"
      @ 12,5 PROMPT "3. ระบบงานเมื่อถูกรองขอ"
      @ 13,5 PROMPT "4. ระบบงานประจํ าภาคเรียน"
      @ 14,5 PROMPT "5. ระบบปรับปรุงและพิมพขอมูลจากแฟม"
      @ 15,5 PROMPT "6. ขอมูลระบบ และทีมพัฒนาโปรแกรม"
      @ 16,5 PROMPT "7. ออกไปสูระบบปฏิบัติการชั่วคราว"
      @ 17,5 PROMPT "8. เลกิการท ํางาน"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <= 5 ;  DO &("SUB"+LTRIM(STR(OPT)))
         CASE OPT = 6 ;  DO WORKSYSTEM
         CASE OPT = 7
              IF MEMORY(2) > 400  ;  RUN COMMAND
              ELSE ; WAIT "หนวยความจํ าไมพอ จึงออกไประบบปฏิบัติการชั่วคราวไมได"
              ENDIF
         CASE OPT = 8 .OR. OPT = 0 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.2
&   ผลลัพธ
// สรางเมนยูอยส ําหรับระบบเตรียมงานและงานลงทะเบียน เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB1
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "1. เมนรูะบบเตรียมงานและงานลงทะเบียน"

      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. รบัขอมูลการเปดวิชาใหม"
      @ 11,5 PROMPT "2. รบัขอมูลขอมูลนักศึกษาใหม"
      @ 12,5 PROMPT "3. รบัลงทะเบียนจากนักศึกษา"
      @ 13,5 PROMPT "4. รบับอกเพิ่ม-ลดวิชา"
      @ 14,5 PROMPT "5. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=4 ;  DO &("R010"+LTRIM(STR(OPT)))
         CASE OPT = 5 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.3
&   ผลลัพธ
// สรางเมนูยอยส ําหรับระบบงานหลังลงทะเบียน เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB2
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "2. เมนูระบบงานหลังลงทะเบียน"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. พมิพขอมูลการลงทะเบียนติดประกาศ"
      @ 11,5 PROMPT "2. พมิพรายชื่อนักเรียนใหอาจารยผูสอน"
      @ 12,5 PROMPT "3. พมิพสรุปการลงทะเบียนแตละวิชาใหผูบริหาร"
      @ 13,5 PROMPT "4. พมิพรายชื่อวิชาที่หองเรียนแตละหองถูกใช"
      @ 14,5 PROMPT "5. พมิพรายชื่อวิชาที่อาจารยผูสอนรับผิดชอบ"
      @ 15,5 PROMPT "6. พมิพสรุปจ ํานวนนักศึกษาแตละวิชา"
      @ 16,5 PROMPT "7. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=6 ;  DO &("R020"+LTRIM(STR(OPT)))
         CASE OPT = 7 ;  EXIT

      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.4
&   ผลลัพธ
// สรางเมนูยอยส ําหรับระบบเมื่อถูกรองขอ เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB3
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "3. เมนูระบบเมื่อถูกรองขอ"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. การขอเพิกถอนวิชา"
      @ 11,5 PROMPT "2. ตรวจสอบตารางเรียนของนักศึกษาเฉพาะคน"
      @ 12,5 PROMPT "3. พมิพรายงานผลการเรียนตลอดหลักสูตร"
      @ 13,5 PROMPT "4. ปรบัปรุงขอมูลอาจารย"
      @ 14,5 PROMPT "5. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=4 ;  DO &("R030"+LTRIM(STR(OPT)))
         CASE OPT = 5 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.5
&   ผลลัพธ
// สรางเมนูยอยส ําหรับระบบงานประจ ําภาคเรียน เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB4
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "4. เมนูระบบงานประจ ําภาคเรียน"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. พมิพรายชื่อติดหนาหองสอบ"
      @ 11,5 PROMPT "2. พมิพรายชื่อแยกตามสถานภาพ"

      @ 12,5 PROMPT "3. พมิพรายชื่อแยกตามสาขา"
      @ 13,5 PROMPT "4. ใสเกรด"
      @ 14,5 PROMPT "5. พมิพรายงานผลการเรียนประจ ําภาค"
      @ 15,5 PROMPT "6. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=5 ;  DO &("R040"+LTRIM(STR(OPT)))
         CASE OPT = 6 ;  EXIT
      ENDCASE
   ENDDO
RETURN
: ตัวอยางที่ 5.6
&   ผลลัพธ
// สรางเมนูยอยสํ าหรับระบบปรับปรุง เพื่อเลือกโปรแกรมตาง ๆ มาทํ างาน
PROCEDURE SUB5
   DO WHILE LASTKEY() != 27
      CLS
      @ 8,5 SAY "5. เมนูระบบปรับปรุงและพิมพขอมูลจากแฟม"
      @ 9,5 SAY "===================="
      @ 10,5 PROMPT "1. แฟมนักศึกษา"
      @ 11,5 PROMPT "2. แฟมวิชา"
      @ 12,5 PROMPT "3. แฟมผูสอน"
      @ 13,5 PROMPT "4. แฟมลงทะเบียน"
      @ 14,5 PROMPT "5. แฟมหอง"
      @ 15,5 PROMPT "6. กลับเมนูหลัก"
      MENU TO OPT
      CLS
      DO CASE
         CASE OPT >= 1 .AND. OPT <=5 ;  DO &("R050"+LTRIM(STR(OPT)))
         CASE OPT = 6 ;  EXIT
      ENDCASE
   ENDDO
RETURN

: ตัวอยางที่ 5.7
&   ผลลัพธ
// พมิพขอมูลของระบบทางจอภาพ เมื่อเลือกตัวเลือกที่ 6 จากเมนูหลัก
// โดยแสดงวันที ่เวลา ชื่อระบบปฏิบัติการ หนวยความจํ าที่เหลือ
// รุนทีใ่ชพัฒนาโปรแกรม สถานะของเครื่องพิมพ ตลอดจนขอมูลของ
// ผูพัฒนาโปรแกรม และสถานที่ติดตอ
PROCEDURE WORKSYSTEM
   CLS
   ? "                วนัที่ :",DTOC(DATE())
   ? "               เวลา :",TIME()
   ? "               วินาที :",SECOND()
   ? "     ระบบปฏิบัติการที่ใช :",OS()
   ? "ขนาดหนวยความจํ าที่เหลือ :",MEMORY(0)
   ? "          รุนของภาษา :",VERSION()
   ? " เครื่องพิมพเปดหรือปดอยู :",ISPRINTER()
   ? "====================================="
   ? "โปรแกรมนี้ถูกพัฒนาโดย. อ.บุรินทร รุจจนพันธุ"
   ? "เริ่มพัฒนาเมื่อ 8 กมุภาพันธ 2545"
   ? "โปรแกรมนี้เปนรุนที ่2.200202"
   ? "หากสนใจ หรือตองการตัวโปรแกรม ติดตอขอรับฟรีไดที"่
   ? "วทิยาลัยโยนก ล ําปาง 52000"
   ? "TEL.(054)265170 EXT.210   FAX. (054)265184"
   INKEY(2)
RETURN

: ตัวอยางที่ 5.8
&   ผลลัพธ
// ตวัอยางนี้ม ี2 ฟงกชันคือ ฟงกชันเพื่อเลือกปลายทางของผลลัพธ
// และฟงกชันถามยํ้ าการปรับปรุงขอมูล เมื่อมีการแกไขขอมูล
// ใชถามวาจะพิมพผลลัพธออกทางใด
FUNCTION ASKPRT()
   ASKPRT = ALERT("ผลลพัธของรายงาน",{'จอภาพ','เครื่องพิมพ'})
RETURN (ASKPRT)
// ใชถามวาจะยอมรับการปรับปรุงขอมูลหรือไม
FUNCTION ASKUPD()
   ASKUPD = ALERT("การปรบัปรุงขอมูล",{'ยอมรับการปรับปรุง','ยกเลิกการปรับปรุง'})
RETURN (ASKUPD)
: ตัวอยางที่ 5.9
&   ผลลัพธ
// เนือ่งจากระบบนี้ไมมีการใชแฟมดรรชน ีแตตองการการเรียงขอมูลใหถูกตอง
// จงึจดัท ําโปรแกรมยอย เพื่อเรียงขอมูลตามฟลดที่ตองการ โดยใชคํ าสั่งจัดเรียง
// ท ําใหผลการจัดเรียงถูกเก็บในแฟมใหม แลวจึงคัดลอกกลับมาทับแฟมเดิม
// ใชจดัเรียงแฟม REGIST ตามป ภาค รหัส วิชา
PROCEDURE SORTREG1
   USE REGIST
   SORT ON RYEAR,RSEM,IDSTD,SUBJ TO TEMPF
   USE TEMPF ; COPY TO REGIST ; CLOSE
RETURN
// ใชจดัเรียงแฟม REGIST ตามวิชา ป ภาค และรหัส
PROCEDURE SORTREG2
   USE REGIST
   SORT ON SUBJ,RYEAR,RSEM,IDSTD TO TEMPF
   USE TEMPF ; COPY TO REGIST ; CLOSE
RETURN
// ใชจดัเรียงแฟม SUBJECT ตามหองเรียน และชื่อวิชา
PROCEDURE SORTSUB1
   USE SUBJECT
   SORT ON ROOM,SUBJ TO  TEMPF
   USE TEMPF ; COPY TO SUBJECT ; CLOSE
RETURN

// ใชจดัเรียงแฟม SUBJECT ตามชื่อวิชา
PROCEDURE SORTSUB2
   USE SUBJECT
   SORT ON SUBJ TO  TEMPF
   USE TEMPF ; COPY TO SUBJECT ; CLOSE
RETURN
// ใชจัดเรียงแฟม IDSTD ตามชื่อรหัส
PROCEDURE SORTID1
   USE IDSTD
   SORT ON IDSTD TO TEMPF
   USE TEMPF ; COPY TO IDSTD ; CLOSE
RETURN
// ใชจดัเรียงแฟม IDSTD ตามชื่อสถานภาพ และรหัส
PROCEDURE SORTID2
   USE IDSTD
   SORT ON ISTAT,IDSTD TO  TEMPF
   USE TEMPF ; COPY TO IDSTD ; CLOSE
RETURN
// ใชจดัเรียงแฟม IDSTD ตามชื่อสาขา และชื่อ
PROCEDURE SORTID3
   USE IDSTD
   SORT ON IMAJOR,INAME,ISURN TO  TEMPF
   USE TEMPF ; COPY TO IDSTD ; CLOSE
RETURN
: ตัวอยางที่ 5.10
&   ผลลัพธ
// ฟงกชนัเปลี่ยนเกรด จากตัวเลข เปนตัวอักษร น ําไปแสดงผลไดชัดเจนขึ้น
FUNCTION  GRADECHR(RGRADE)
   GRADECHR = '--'
   DO CASE
      CASE RGRADE = 4    ;  GRADECHR = 'A '
      CASE RGRADE = 3.5  ;  GRADECHR = 'B+'
      CASE RGRADE = 3    ;  GRADECHR = 'B '
      CASE RGRADE = 2.5  ;  GRADECHR = 'C+'

      CASE RGRADE = 2    ;  GRADECHR = 'C '
      CASE RGRADE = 1.5  ;  GRADECHR = 'D+'
      CASE RGRADE = 1    ;  GRADECHR = 'D '
      CASE RGRADE = 0    ;  GRADECHR = 'F '
      CASE RGRADE = 4.2  ;  GRADECHR = 'W '
      CASE RGRADE = 4.6  ;  GRADECHR = 'P '
   ENDCASE
RETURN (GRADECHR)
: ตัวอยางที่ 5.11
&   ผลลัพธ
// ฟงกชนัปลี่ยนเกรด จากตัวอักษรเปนตัวเลข ท ําใหจัดเก็บลงแฟมไดงายขึ้น
FUNCTION  GRADENUM(_GRADE)
   GRADENUM = 0
   DO CASE
      CASE UPPER(_GRADE) = 'A'  ; GRADENUM = 4
      CASE UPPER(_GRADE) = 'B+' ; GRADENUM = 3.5
      CASE UPPER(_GRADE) = 'B'  ; GRADENUM = 3
      CASE UPPER(_GRADE) = 'C+' ; GRADENUM = 2.5
      CASE UPPER(_GRADE) = 'C'  ; GRADENUM = 2
      CASE UPPER(_GRADE) = 'D+' ; GRADENUM = 1.5
      CASE UPPER(_GRADE) = 'D'  ; GRADENUM = 1
      CASE UPPER(_GRADE) = 'F'  ; GRADENUM = 0
      CASE UPPER(_GRADE) = 'W'  ; GRADENUM = 4.2
      CASE UPPER(_GRADE) = 'P'  ; GRADENUM = 4.6
   ENDCASE
RETURN (GRADENUM)
: ตัวอยางที่ 5.12
&   ผลลัพธ
// แสดงขอความงาย ๆ ทันท ีเมื่อกดปุมฟงกชัน F1 โดยไมท ําใหภาพเดิมบนจอเปลี่ยน
// เพราะมกีารใชค ําสั่งเก็บสถานะของจอ และคืนสถานะเดิมของจอ และที่ใช F1 ได
// เพราะ ระบุในโปรแกรมหลัก MAINALL วา SET KEY 28 TO HELPFORM
// แสดงขอมูลของผูพัฒนาโปรแกรมใหทราบ

PROCEDURE HELPFORM
   SAVE SCREEN TO SCR
   @ 5,5 CLEAR TO 20,75 // ลบขอความบนพื้นที่กํ าหนด
   @ 9,8 TO 15,71  // ตกีรอบรอบขอความ
   @ 10,10 SAY "CLIPPER PROGRAMING"
   @ 11,10 SAY "LEARING BY CASE STUDY"
   @ 12,10 SAY "GOOD LUCK"
   @ 13,10 SAY DATE()
   @ 14,10 SAY TIME()
   INKEY(5)
   RESTORE SCREEN FROM SCR
RETURN
: ตัวอยางที่ 5.13
&   ผลลัพธ
// ฟงกชนันีช้วยใหการใสคาในตัวแปรบางตัวแปรถูกตอง เพราะจะแสดง
// คาทีเ่หมาะสมกับฟลดนั้นมาใหเลือก โดยการอานคาจากแฟมมาแสดงให
// ด ูแตจะตองระบุใหชัดเจนวาจะน ําฟลดใดมาใชบาง
// ฟงกชันนี้จึงแสดงขอมูลที่เปนไปไดส ําหรับตัวแปรที่กํ าลังรอรับคา
FUNCTION HELPREADVAR
   OLDALIAS = ALIAS() // เก็บพื้นที่ท ํางานเดิม
   SHOWAR = {}
   DO CASE
      CASE READVAR() = "_SUBJ"
           IF OLDALIAS != "SUBJECT"
              USE SUBJECT NEW // ถาเปดแลวตองไมเปดอีก
           ENDIF
           GO TOP
           WHILE !EOF()
              AADD(SHOWAR,SUBJ)
              SKIP
           END
           IND := ACHOICE(2,71,5,78,SHOWAR)
           IF IND > 0 // ปองกันการใชปุม ESC
              _SUBJ = SHOWAR[IND]

           ENDIF
           IF OLDALIAS != "SUBJECT" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
              CLOSE                 // ในจุดที่ไมไดเปดแฟมนี้
           ENDIF
      CASE READVAR() = "_IDSTD"
           IF OLDALIAS != "IDSTD"
              USE IDSTD NEW  // ถาเปดแลวตองไมเปดอีก
           ENDIF
           GO TOP
           FOR I = 1 TO  RECCOUNT()
               AADD(SHOWAR,STR(I,2)+". "+STR(IDSTD)+" : "+INAME)
               SKIP
           NEXT
           IND := ACHOICE(2,50,10,78,SHOWAR)
           IF IND > 0  // ปองกันการใชปุม ESC
              _IDSTD = VAL(SUBSTR(SHOWAR[IND],5,7))
           ENDIF
           IF OLDALIAS != "IDSTD" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
              CLOSE               // ในจุดที่ไมไดเปดแฟมนี้
           ENDIF
       OTHERWISE
           X = ALERT("ขอโทษ ไมไดเตรียมสวนสนับสนุน ส ําหรับจุดนี้",{"OK."})
   ENDCASE
   IF LEN(OLDALIAS) > 0
      SELE &OLDALIAS // กลับสูพื้นที่ท ํางานเดิม
   ENDIF
RETURN

: ตัวอยางที่ 5.14
&   ผลลัพธ
// สอบถามการเลิกท ํางานดวยฟงกชัน ALERT
PROCEDURE QUITPROC
   Q =  ALERT("สอบถามการเลิกการท ํางาน",{"เลิกการทํ างาน","ยังไมเลิกงาน"})
   IF Q = 1
      QUIT
   ENDIF
RETURN

5.4.2 โปรแกรมตาง ๆ ใน SUB01.PRG
โปรแกรมใน SUB01.PRG นี้มีโปรแกรมยอยทั้งหมด 4 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบเตรียม

งาน และงานลงทะเบียน เชน โปรแกรมรับขอมูลการลงทะเบียน โปรแกรมขอบอกเพิ่มลดวิชา โปรแกรมปอนขอมูลนัก
ศกึษา หรือโปรแกรมปรับปรุงรายวิชาเปนตน

ระบบเตรยีมงาน และงานลงทะเบียนเปนระบบที่ส ําคัญยิ่ง เพราะระบบนี้ถูกพัฒนาขึ้นเพื่อเก็บประวัติการ
เรยีน และโปรแกรมในระบบนี้จะบันทึกขอมูลการลงทะเบียนของนักศึกษาแตละคน  ท ําใหเกิดขอมูลขึ้น เสมือนวาจุดนี้
เปนจดุเริ่มตนที่ท ําใหเกิดขอมูล  น ําไปปรับปรุง น ําไปใช และประมวลผลในงานของระบบอื่นไดอีกมาก

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB01.PRG ทั้งหมด 4 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.15 ถึง ตัวอยางที่ 5.18 ( R0101 - R0104 )
: ตัวอยางที่ 5.15
&   ผลลัพธ
// ปรบัปรงุขอมูลในแฟมวิชาอยางงาย ๆ ดวยฟงกชัน BROWSE
// แลวสอบถามวาจะปรับปรุงหรือไม ถาไมจะยกเลิกการปรับปรุงทั้งหมด
PROCEDURE R0101
   USE SUBJECT
   COPY TO TEMPF
   BROWSE(5,5,22,75)
   IF ASKUPD() = 1
      PACK
   ELSE
      USE TEMPF
      COPY TO SUBJECT
   ENDIF
   CLOSE ALL
RETURN

: ตัวอยางที่ 5.16
&   ผลลัพธ
// เพิม่รายชือ่นักศึกษาใหม เพียงคนเดียว โดยระบุรหัสใหมใหอัตโนมัติ
PROCEDURE R0102
   USE IDSTD
   GOTO BOTTOM
   _IDSTD = IDSTD + 1
   _INAME = SPACE(20)
   _ISURN = SPACE(20)
   _IMAJOR = 0
   _ISTAT = 'A'
   @ 10,5 SAY "เพิม่รายชื่อนักศึกษาใหม"
   @ 11,5 SAY "รหัส"+STR(_IDSTD)
   @ 12,5 SAY "ชื่อ" GET _INAME
   @ 13,5 SAY "สกุล" GET _ISURN
   @ 14,5 SAY "สาขา" GET _IMAJOR
   @ 15,5 SAY "สถานภาพ" GET _ISTAT
   READ
   IF ASKUPD() = 1 // สอบถามการปรับปรุง
      APPEND BLANK
      REPLACE IDSTD WITH _IDSTD ,INAME WITH _INAME ,ISURN WITH _ISURN,;
              IMAJOR WITH _IMAJOR,ISTAT WITH _ISTAT
   ENDIF
RETURN
: ตัวอยางที่ 5.17
&   ผลลัพธ
// เพิม่ขอมูลการลงทะเบียนของนักศึกษาแตละคน
// วธินีีจ้ะลงทะเบียนและตัดยอดใหทันท ีส ําหรับแตละวิชา
// หากลง 5 วิชาแตผิดพลาด 2 วิชา จะเพิ่มขอมูลเขาไป 3 วิชาทันที
// การปรบัปรงุจะเพิ่มจ ํานวนอีก 1 เขาไปในฟลดจํ านวนที่ลง ในแฟมวิชาแตละวิชา
// และเพิม่เรคอรดเทาจ ํานวนวิชาที่ลงทะเบียนไดในแฟมลงทะเบียน
// พมิพใบเสร็จโดยคิดคาลงทะเบียนเรียนวิชาละ 1000 บาท
PROCEDURE R0103
   SET CENTURY ON

   _RYEAR = IIF(MONTH(DATE())<5,YEAR(DATE())-1,YEAR(DATE()))
   _RSEM = '1'
   _IDSTD = 0
   _SUBJ := ARRAY(8)
   _RGRADE = 4.6
   @ 4,5 SAY "โปรแกรม ลงทะเบียนเรียน"
   @ 5,5 SAY "ประจ ําปการศึกษา : " GET _RYEAR PICT '9999'
   @ 6,5 SAY "   ภาคการศึกษา : " GET _RSEM
   SAVE SCREEN TO SCR
   DO WHILE LASTKEY() != 27
      AFILL(_SUBJ,SPACE(8))  // ก ําหนดคาเริ่มตนใหตัวแปรอาเรย _SUBJ ทุกสมาชิก
      _IDSTD = 0
      DO WHILE .T. // รบัขอมูลนักศึกษาที่มีในแฟม IDSTD เทานั้น
         USE IDSTD
         @ 7,5 SAY "    รหสันักศึกษา : " GET _IDSTD PICT '9999999'
         READ
         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            _INAME = INAME
            _ISURN = ISURN
            @ 8,5 SAY INAME+' '+ISURN+' '+STR(IMAJOR)+' '+ISTAT
            EXIT
         ELSE ; @ 8,5 SAY 'ไมมีรหัสนักศึกษานี้ในแฟม IDSTD'
         ENDIF
      ENDDO
      @ 10,10 SAY "SUBJ1 : " GET _SUBJ[1] PICT '@!'
      @ 11,10 SAY "SUBJ2 : " GET _SUBJ[2] PICT '@!'
      @ 12,10 SAY "SUBJ3 : " GET _SUBJ[3] PICT '@!'
      @ 13,10 SAY "SUBJ4 : " GET _SUBJ[4] PICT '@!'
      @ 14,10 SAY "SUBJ5 : " GET _SUBJ[5] PICT '@!'
      @ 15,10 SAY "SUBJ6 : " GET _SUBJ[6] PICT '@!'
      @ 16,10 SAY "SUBJ7 : " GET _SUBJ[7] PICT '@!'
      @ 17,10 SAY "SUBJ8 : " GET _SUBJ[8] PICT '@!'

      READ
      YOURSUBJ = 8  // จ ําเปนตอง 8 เพราะตรวจชองวางของวิชา
      OKSUBJ = 0
      USE SUBJECT
      FOR I = 1 TO 8 // น ําวชิามาตรวจสอบในแฟมวิชา แลวเพิ่มจํ านวนเขาไป
         IF _SUBJ[I] != SPACE(8)
            LOCATE FOR _SUBJ[I] = SUBJ
            IF FOUND()
               IF SREGIST >= SLIMIT
                  @ 9+I,40 SAY 'วชิานีเ้ตม็แลว เพราะรับเพียง'+STR(SLIMIT)
               ELSE
                  @ 9+I,40 SAY 'OK.' // วชิาลงเพิ่ม จึงเพิ่มคาในฟลดจํ านวนที่ลง
                  REPLACE SREGIST WITH SREGIST + 1
                  USE REGIST          // เพิม่เรคอรดเขาไปในแฟม REGIST
                     APPEND BLANK       // 4.6 คอืเกรด P หมายถึงกํ าลังศึกษาอยู
                     REPLACE RYEAR WITH _RYEAR, RSEM WITH _RSEM ,;
                        IDSTD WITH _IDSTD, SUBJ WITH _SUBJ[I],RGRADE WITH 4.6
                  USE SUBJECT
                  OKSUBJ = OKSUBJ + 1
               ENDIF
            ELSE
               @ 9+I,40 SAY 'ไมมีวิชานี้ในแฟม SUBJECT'
            ENDIF
         ELSE
            YOURSUBJ = I - 1
            I = 8
         ENDIF
      NEXT
      @ 18,10 SAY 'จ ํานวนวชิาที่ทานลงทะเบียนเรียนคือ '+STR(YOURSUBJ)
      @ 19,10 SAY 'จ ํานวนวิชาที่ทานลงไดคือ '+STR(OKSUBJ)
      INKEY(2)
      // สวนนี้ใชพิมพใบเสร็จ โดยคิดวิชาละ 1000 บาท
      IF OKSUBJ >= 1

         GASKPRT = ASKPRT()
         IF GASKPRT = 2 .AND. ISPRINTER()  // เลข 2 หมายถึงเครื่องพิมพ
            SET DEVICE TO PRINTER
         ELSE ; CLS
         ENDIF
         @ 1,5  SAY "รหัส :"+STR(_IDSTD)
         @ 2,5  SAY "ชื่อ-สกุล :"+_INAME+" "+_ISURN
         @ 3,5  SAY "==========================="
         FOR I = 1 TO YOURSUBJ
            @ 3+I,5 SAY _SUBJ[I]
         NEXT
         @ 4+YOURSUBJ,5 SAY "จ ํานวนเงนิทั้งหมดที่ทานตองช ําระ"+STR(OKSUBJ*1000)
         IF GASKPRT = 2 .AND. ISPRINTER() ; EJECT ; ENDIF
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(0) // ขณะรอกดปุม ถากด ESC จะเลิกงานลงทะเบียน ถาไมจะรับคนใหม
      RESTORE SCREEN FROM SCR
   ENDDO
   SET CENTURY OFF
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.18
&   ผลลัพธ
// บอกเพิ่ม-ลดวิชา เพื่อปรับปรุงขอมูลการลงทะเบียน
// การปรบัปรงุส ําหรับการบอกเพิ่มคือเพิ่มจํ านวนในฟลดจํ านวนที่ลง แตบอกเลิกจะลดลง
// และจะเพิ่มหรือลดเรคอรดเทาจ ํานวนที่ระบุในแฟมลงทะเบียน
// สดุทายพมิพใบเสร็จ โดยคิดวิชาละ 1000 บาท และคาธรรมเนียมวิชาละ 20 บาท
PROCEDURE R0104
   SET CENTURY ON
   _RYEAR = IIF(MONTH(DATE())<5,YEAR(DATE())-1,YEAR(DATE()))
   _RSEM = '1'
   _IDSTD = 0
   _SUBJ := ARRAY(8)
   @ 4,5 SAY "โปรแกรม บอกเพิ่ม-ลดวิชา"

   @ 5,5 SAY "ประจํ าปการศึกษา : " GET _RYEAR
   @ 6,5 SAY "   ภาคการศึกษา : " GET _RSEM
   SAVE SCREEN TO SCR
   DO WHILE LASTKEY() != 27
      AFILL(_SUBJ,SPACE(8))  // ก ําหนดคาเริ่มตนใหตัวแปรอาเรย _SUBJ ทุกสมาชิก
      _IDSTD = 0
      @ 7,5 SAY "    รหสันักศึกษา : " GET _IDSTD PICT '9999999'
      READ
      @ 8,10 SAY "เพิ่มวิชา"
      @ 9,10 SAY "SUBJ1 : " GET _SUBJ[1] PICT '@!'
      @ 10,10 SAY "SUBJ2 : " GET _SUBJ[2] PICT '@!'
      @ 11,10 SAY "SUBJ3 : " GET _SUBJ[3] PICT '@!'
      @ 12,10 SAY "SUBJ4 : " GET _SUBJ[4] PICT '@!'
      @ 13,10 SAY "ลดวิชา"
      @ 14,10 SAY "SUBJ5 : " GET _SUBJ[5] PICT '@!'
      @ 15,10 SAY "SUBJ6 : " GET _SUBJ[6] PICT '@!'
      @ 16,10 SAY "SUBJ7 : " GET _SUBJ[7] PICT '@!'
      @ 17,10 SAY "SUBJ8 : " GET _SUBJ[8] PICT '@!'
      READ
      YOURADD := YOURDROP := 4
      OKSUBJADD := OKSUBJDROP := 0
      USE SUBJECT
      FOR I = 1 TO 4
         IF _SUBJ[I] != SPACE(8)
            LOCATE FOR _SUBJ[I] = SUBJ
            IF FOUND()
               IF SREGIST >= SLIMIT
                  @ 8+I,40 SAY 'วชิานีเ้ตม็แลว เพราะรับเพียง'+STR(SLIMIT)
               ELSE
                  @ 8+I,40 SAY 'OK.' // วชิาลงเพิ่ม จึงเพิ่มคาในฟลดจํ านวนที่ลง
                  REPLACE SREGIST WITH SREGIST + 1
                  USE REGIST
                     APPEND BLANK      // 4.6 คอืเกรด P หมายถึงกํ าลังศึกษาอยู

                     REPLACE RYEAR WITH _RYEAR, RSEM WITH _RSEM ,;
                        IDSTD WITH _IDSTD, SUBJ WITH _SUBJ[I],RGRADE WITH 4.6
                  USE SUBJECT
                  OKSUBJADD = OKSUBJADD + 1
               ENDIF
            ENDIF
         ELSE
            YOURADD = I - 1   ; I = 4
         ENDIF
      NEXT
      USE REGIST  // การบอกเลิกจะตองมีเรคอรดที่นั้นกอนจึงจะบอกเลิกได
      FOR I = 5 TO 8
         IF _SUBJ[I] != SPACE(8)
            LOCATE FOR _RYEAR = RYEAR .AND. _RSEM = RSEM .AND. ;
                       _IDSTD = IDSTD .AND. _SUBJ[I] = SUBJ
            IF FOUND()
               DELETE
               USE SUBJECT
               LOCATE FOR _SUBJ[I] = SUBJ
               IF FOUND()
                  @ 9+I,40 SAY 'OK.' // วชิานีม้คีนบอกเลิก จึงลดคาใน SREGIST
                  REPLACE SREGIST WITH SREGIST - 1
                  OKSUBJDROP = OKSUBJDROP + 1
               ENDIF
               USE REGIST
            ENDIF
         ELSE
            YOURDROP = I - 5   ; I = 8
         ENDIF
      NEXT
      PACK  // ลบเรคอรดที่ท ําเครื่องหมายลบไว ออกจากแฟมอยางแทจริง
      @ 18,10 SAY 'จ ํานวนวชิาที่ทานบอกเพิ่ม '+STR(YOURADD)
      @ 19,10 SAY 'จ ํานวนวชิาที่ทานบอกเลิก '+STR(YOURDROP)

      @ 20,10 SAY 'จ ํานวนวิชาที่บอกเพิ่มได '+STR(OKSUBJADD)
      @ 21,10 SAY 'จ ํานวนวชิาที่บอกเลิกได '+STR(OKSUBJDROP)
      INKEY(2)
      // พมิพใบเสร็จ คิดวิชาละ 1000 บาท และคาธรรมเนียมเพิ่มลด วิชาละ 20 บาท
      IF (OKSUBJADD + OKSUBJDROP) >= 1
         GASKPRT = ASKPRT()
         IF GASKPRT = 2 .AND. ISPRINTER()  // เลข 2 หมายถึงเครื่องพิมพ
            SET DEVICE TO PRINTER
         ELSE ; CLS
         ENDIF
         @ 1,5  SAY "รหัส :"+STR(_IDSTD)
         @ 2,5  SAY "==========================="
         FOR I = 1 TO YOURADD  // พมิพวิชาที่บอกเพิ่ม
            @ 3+I,5 SAY 'A : '+ _SUBJ[I]
         NEXT
         FOR I = 1 TO YOURDROP // พมิพวิชาที่บอกเลิก
            @ 3+I+YOURADD,5 SAY 'D : '+ _SUBJ[I+4]
         NEXT
         @ 4+YOURADD+YOURDROP,5 SAY "จ ํานวนเงินทั้งหมดที่ทานตองชํ าระ หรือรับคืน"+ ;
               STR(((OKSUBJADD-OKSUBJDROP)*1000)+(OKSUBJADD+OKSUBJDROP)*20)
         IF GASKPRT = 2 .AND. ISPRINTER() ; EJECT ; ENDIF
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(0) // ขณะรอกดปุม ถากด ESC จะเลิกงานลงทะเบียน ถาไมจะรับคนใหม
      RESTORE SCREEN FROM SCR
   ENDDO
   SET CENTURY OFF
RETURN

5.4.3 โปรแกรมตาง ๆ ใน SUB02.PRG
โปรแกรมใน SUB02.PRG นี้มีโปรแกรมยอยทั้งหมด 6 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบงาน หลัง

ลงทะเบยีน เชน โปรแกรมพิมพรายชื่อนักศึกษาที่ลงทะเบียนติดประกาศ  พิมพรายชื่อใหอาจารยผูสอน  พิมพรายชื่อ
วชิาตดิหนาหองเรียน  พิมพสรุปรายงานใหผูบริหาร  เปนตน

ระบบงานหลังลงทะเบียนเปนระบบที่ท ําใหเกิดประสิทธิภาพในการเรียนการสอน เพราะอาจารยผูสอนจะได
รบัรายชือ่ เพือ่ตรวจสอบเวลาเรียนและลงคะแนน ผูบริหารทราบจ ํานวนนักศึกษาที่ลงทะเบียน ทราบสถิติการใชหอง
เรยีนวามปีระสทิธิภาพเพียงใด  ดังนั้นระบบนี้จึงมุงที่จะทํ ารายงานใหหนวยงานตาง ๆ หลังลงทะเบียนทันที

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB02.PRG ทั้งหมด 6 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.19 ถึง ตัวอยางที่ 5.24 ( R0201 - R0206 )
: ตัวอยางที่ 5.19
&   ผลลัพธ
// พมิพขอมูลการลงทะเบียนเรียนติดประกาศ ใหนักศึกษาดู
// รายงานแสดงล ําดับ รหัสนักศึกษา และรหัสวิชา
// เมือ่เริม่โปรแกรมจะท ําการจัดเรียงขอมูลใหมหมด แลวแยกกลุมขอมูล โดยใชรหัส
// จะน ําขอมูลเฉพาะป 1996 ภาคเรียนที่ 1 เทานั้นมาแสดง
// ในขณะพมิพขอมูลสามารถนับจ ํานวนของนักศึกษา และวิชาของนักศึกษาแตละคน
PROCEDURE R0201
   DO SORTREG1 // ใชจดัเรียงขอมูลตามป ภาค รหัส และวิชา อยูใน MAIN.PRG
   USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
   GO TOP
   _IDSTDOLD = 0
   IIDSTD := ISUBJ := 0
   DO WHILE !EOF()
      IF IDSTD <> _IDSTDOLD
        IIDSTD++
        ISUBJ = 0
        INKEY(0.5)
        ? '* --> ',IIDSTD,IDSTD
        _IDSTDOLD = IDSTD
      ENDIF
      ISUBJ++
      ? ISUBJ,SUBJ
      SKIP
   ENDDO
   INKEY(0)
   CLOSE ALL
RETURN

: ตัวอยางที่ 5.20
&   ผลลัพธ
// พมิพรายชื่อนักศึกษาใหอาจารยผูสอน
// รายงานแสดงล ําดับ รหัสวิชา รหัสนักศึกษา ชื่อ สกุล รหัสสาขา และสถานภาพ
// เมือ่เริม่โปรแกรมตองจัดเรียงขอมูลใหเรียงตามรหัสวิชากอน
PROCEDURE R0202
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG
   SELE 1
   USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
   GO TOP
   SELE 2
   USE IDSTD
   SELE 1
   _SUBJOLD = SPACE(8)
   IIDSTD := ISUBJ := 0
   DO WHILE !EOF()
      IF SUBJ <> _SUBJOLD
         ISUBJ++
         IIDSTD = 0
         INKEY(0.5)
         ? '* --> ',ISUBJ,SUBJ
         _SUBJOLD = SUBJ
      ENDIF
      IIDSTD++
      _IDSTD = IDSTD
      SELE 2
      LOCATE FOR IDSTD = _IDSTD
      IF FOUND()
         ? IIDSTD,IDSTD,INAME,ISURN,IMAJOR,ISTAT
      ELSE ; ? IIDSTD,_IDSTD  // เพราะรหัสนี้ไมมีใน IDSTD จึงตองใชตัวแปรแทน
      ENDIF
      SELE 1 // ตองมาที่พื้นที ่1 จึงจะ SKIP เพราะตองการเลื่อนเรคอรดใน REGIST
      SKIP

   ENDDO
   INKEY(0)
   CLOSE ALL // ตอง CLOSE ALL เพราะเปดแฟมไวหลายพื้นที่
RETURN
: ตัวอยางที่ 5.21
&   ผลลัพธ
// พมิพสรุปการลงทะเบียนใหผูบริหาร โดยพิมพจํ านวนที่ลงเรียนในแตละวิชา
// โปรแกรมนี้คุมกลุม 2 แบบคือ กลุมของวิชา และกลุมของภาคการศึกษา
// รายงานแสดงปการศึกษา ภาคการศึกษา วิชา จํ านวนที่ลงทะเบียนเรียน
// สามารเลือกปลายทางของผลลัพธได
// ควบคมุจ ํานวนบรรทัดตอหนาของจอภาพเปน 20 และของเครื่องพิมพเปน 30
PROCEDURE R0203
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG
   USE REGIST
   _SUBJO = SUBJ
   _RYEARO = RYEAR
   _RSEMO = RSEM
   IIDSTD := ISUBJ := 0
   GASKPRT = ASKPRT()
   IF GASKPRT = 2 // เลข 2 หมายถึงพิมพรายงานทางเครื่องพิมพ
      SET DEVICE TO PRINTER
      CNTPAGE = 30
   ELSE
      CNTPAGE = 20
   ENDIF
   CNTLINE = 1
   DO WHILE !EOF()
      IIDSTD++
      IF SUBJ <> _SUBJO .OR. ISUBJ = 0   // ตรวจสอบความตางของวิชา
         INKEY(0.5)
         ISUBJ++
         @ CNTLINE,5 SAY STR(ISUBJ)+SUBJ
         _SUBJO = SUBJ
         CNTLINE++

      ENDIF
      SKIP           // ตรวจสอบภาคการศึกษา
      IF SUBJ <> _SUBJO .OR. _RYEARO <> RYEAR .OR. _RSEMO <> RSEM .OR. EOF()
         @ CNTLINE,10 SAY STR(_RYEARO) + _RSEMO + STR(IIDSTD)
         CNTLINE++
         _RYEARO = RYEAR
         _RSEMO = RSEM
         IIDSTD = 0
      ENDIF
      IF CNTLINE >= CNTPAGE .OR. EOF() // ตรวจสอบการแบงหนา
         CNTLINE = 1
         IF GASKPRT = 2 ; EJECT
         ELSE ; INKEY(3) ; CLS ; ENDIF
      ENDIF
   ENDDO
   SET DEVICE TO SCREEN
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.22
&   ผลลัพธ
// พมิพรายชื่อวิชาของแตละหอง โดยเลือกปลายทางของผลลัพธได
// รายงานแสดงล ําดับและรหัสหอง รหัสวิชา เวลาเรียน และจ ํานวนที่ลงทะเบียน
// โปรแกรมนี้จะแบงหองเรียน แลวรายงานชื่อวิชาในหองนั้น
// โดยรายงานจะแสดงหองละ 1 หนาเทานั้น
PROCEDURE R0204
   DO SORTSUB1   // ใชจดัเรยีงแฟมวิชาตามหองเรียนและชื่อวิชา
   USE SUBJECT
   _ROOMO = SPACE(4)
   IROOM := ISUBJ := 0
   GASKPRT = ASKPRT()
   IF GASKPRT = 2
      SET DEVICE  TO PRINTER
   ENDIF

   DO WHILE !EOF()
      IF ROOM <> _ROOMO
         IF GASKPRT = 2 .AND. IROOM != 0
            EJECT
         ELSE
            INKEY(3)
            CLS
         ENDIF
         IROOM++
         @ 1,1 SAY "รายวชิาที่มีการใชในแตละหอง"
         @ 2,1 SAY "หองที ่"+STR(IROOM)+" เบอร "+ROOM
         @ 3,1 SAY "============================"
         _ROOMO = ROOM
         ISUBJ = 0
      ENDIF
      ISUBJ++
      @ ISUBJ+3,1 SAY STR(ISUBJ) + SUBJ + STIME + STR(SREGIST)
      SKIP
   ENDDO
   SET DEVICE TO SCREEN
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.23
&   ผลลัพธ
// พมิพรายชื่อวิชาที่อาจารยผูสอนแตละทานรับผิดชอบ
// รายงานแสดงรหัสอาจารย ชื่ออาจารย รหัสวิชา หองเรียน และเวลาเรียน
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE R0205
   DO SORTSUB2
   SELE 1
   USE TEACHER
   SELE 2
   USE SUBJECT

   IF ASKPRT() = 2  // ไดผลลัพธออกเครื่องพิมพ
      SET DEVICE TO PRINTER
      FOR I= 1 TO RECCOUNT()
          @ I,10 SAY STR(I) + SUBJ + ' | ' + ROOM + ' | ' + STIME
          SELE 1
             LOCATE FOR SUBJECT->TEACHID = TEACHID
             IF FOUND()
                @ I,50 SAY ' | ' + TNAME + ' | ' + TSURN
             ENDIF
             IF MOD(I,15) = 0
                EJECT
             ENDIF
          SELE 2
          SKIP
      NEXT
      EJECT
      SET DEVICE TO SCREEN
   ELSE
      I = 0  // ไดผลลัพธออกทางจอภาพ
      DO WHILE !EOF()
         I++
         ? I,SUBJ,ROOM,STIME
         SELE 1
            LOCATE FOR SUBJECT->TEACHID = TEACHID
            IF FOUND() ; ?? TNAME,TSURN ; ENDIF
            IF I = 15
               INKEY(3)
               CLS
               I = 0
            ENDIF
         SELE 2
         SKIP
      ENDDO

   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.24
&   ผลลัพธ
// พมิพสรุปจ ํานวนนักศึกษาแตละวิชา โดยระบุป และภาค
// รายงานแสดงล ําดับ รหัสวิชา จํ านวนที่ลงทะเบียน
// สามารถเลือกรูปแบบของผลลัพธได
PROCEDURE R0206
   _RYEAR = 0
   _RSEM = '0'
   @ 4,5 SAY 'สรปุจ ํานวนนักศึกษาในปและภาคที่ระบุ'
   @ 5,5 SAY 'ปการศึกษา : ' GET _RYEAR
   @ 6,5 SAY 'ภาคการศึกษา : ' GET _RSEM
   READ
   CLS
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG
   USE REGIST
   SET FILTER TO RYEAR = _RYEAR .AND. RSEM = _RSEM
   GO TOP
   _SUBJO = SUBJ
   IIDSTD := ISUBJ := 0
   GASKPRT = ASKPRT()
   CNTLINE = 1
   CNTPAGE = 20 // เริ่มตนส ําหรับจอภาพใหหนาละ 20 บรรทัด
   IF GASKPRT = 2 // เลข 2 หมายถึงพิมพรายงานทางเครื่องพิมพ
      SET DEVICE TO PRINTER
      CNTPAGE = 30
   ENDIF
   @ CNTLINE,5 SAY "สรปุจ ํานวนนักศึกษาประจํ าป " + STR(_RYEAR) + "/" + _RSEM
   CNTLINE++
   DO WHILE !EOF()
      IIDSTD++

      SKIP
      IF SUBJ <> _SUBJO .OR. EOF()   // ตรวจสอบความตางกลุมของวิชา
         ISUBJ++
         @ CNTLINE,5 SAY STR(ISUBJ)+_SUBJO+STR(IIDSTD)
         _SUBJO = SUBJ
         CNTLINE++
         IIDSTD = 0
      ENDIF
      IF CNTLINE >= CNTPAGE .OR. EOF()  // ตรวจสอบการแบงหนา
         CNTLINE = 1
         IF GASKPRT = 2 ; EJECT
         ELSE ; INKEY(3) ; CLS ; ENDIF
      ENDIF
   ENDDO
   SET DEVICE TO SCREEN
   INKEY(3)
   CLOSE ALL
RETURN

5.4.4 โปรแกรมตาง ๆ ใน SUB03.PRG
โปรแกรมใน SUB03.PRG นี้มีโปรแกรมยอยทั้งหมด 4 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบงาน เมื่อ

ถกูรองขอ เชน โปรแกรมขอถอนวิชาเรียน  โปรแกรมพิมพตารางเรียน โปรแกรมพิมพใบรายงานผลการเรียนโปรแกรม
ปรบัปรุงขอมูลอาจารย เปนตน

ระบบงานเมือ่ถูกรองขอ เปนระบบเกิดขึ้นเมื่อมีผูรองขอปรับปรุง หรือรายงานจากระบบ โดยไมขึ้นกับเวลา วา
จะเปนตนภาคเรียน ทายภาค หรือชวงปดภาคเรียน  แตในระบบอื่น ๆ ผูจัดระบบอาจแยกโปรแกรมตาง ๆ ในระบบนี้เขา
ไปรวมกบัระบบอื่นได  แตผูเขียนแยกออกมาเพื่อใหเห็นการแยกระบบยอย ในอีกมุมมองหนึ่ง

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB03.PRG ทั้งหมด 4 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.25 ถึง ตัวอยางที่ 5.28 ( R0301 - R0304 )
: ตัวอยางที่ 5.25
&   ผลลัพธ
// ปอนขอมลูการขอถอนวิชาของนักศึกษา (W) โดยระบุรหัสนักศึกษา และวิชา
// ส ําหรบัปการศึกษาและภาคเรียนจะระบุไวแนนอนในโปรแกรมเปนป 1996 ภาคเรียนที่ 1
// โปรแกรมจะคนหาขอมูลแลวเปลี่ยนเกรด จากนั้นจะรอรับรหัสนักศึกษาและวิชาใหม
// จนกวาจะกดปุม ESC เพื่อเลิกการทํ างาน

PROCEDURE R0301
   USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM= '1'
   DO WHILE LASTKEY() != 27
      GO TOP
      _IDSTD = 0
      _SUBJ = SPACE(8)
      @ 6,5 CLEAR
      @ 4,5 SAY "รหสันักศึกษาที่ขอเพิกถอนวิชา : " GET _IDSTD
      @ 5,5 SAY "รหสัวิชาที่ขอเพิกถอนวิชา : " GET _SUBJ
      READ
      LOCATE FOR IDSTD = _IDSTD .AND. SUBJ = _SUBJ
      IF FOUND()
         @ 6,5 SAY 'เกรดเดิม ' + STR(RGRADE)
         @ 7,5 SAY 'การขอเพิกถอนของ ' + STR(_IDSTD) + ' สมบูรณ'
         REPLACE RGRADE WITH 4.2
      ELSE
         @ 6,5 SAY 'ไมพบรหัสนักศึกษา ' + STR(_IDSTD)
      ENDIF
      INKEY(0)
   ENDDO
RETURN
: ตัวอยางที่ 5.26
&   ผลลัพธ
// พมิพตารางเรียนของนักศึกษาแตละวิชา
// รายงานแสดงใหเห็นวานักศึกษาแตละคน เรียนวิชาอะไร หอง เวลาและใครคือผูสอน
// ตวัอยางนี้แสดงการใช DEVPOS และ DEVOUT เพื่อน ําขอมูลมาแสดง
// เปนอีกวิธีหนึ่งที่จะน ําขอมูลออกทางเครื่องพิมพได
// แสดงใหเห็นการเชื่อมแฟมจาก 3 แฟมคือ TEACHER,SUBJECT และ REGIST
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE R0302
   DO SORTREG1   // จดัเรียงตามป ภาค และรหัส
   SELE 1 ; USE TEACHER
   SELE 2 ; USE SUBJECT

   SELE 3 ; USE REGIST
   SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
   GO TOP
   _IDSTDO = 0
   GASKPRT = ASKPRT()
   IF GASKPRT = 2
      SET DEVICE TO PRINTER
   ENDIF
   DO WHILE !EOF()
      IF IDSTD <> _IDSTDO
         IF GASKPRT = 2 ;  DEVPOS(PROW()+1,1)
            ELSE ; DEVPOS(ROW()+1,1) ; ENDIF
         DEVOUT(IDSTD)
         _IDSTDO = IDSTD
      ENDIF
      SELE 2 // แฟม SUBJECT.DBF
      LOCATE FOR SUBJ = REGIST->SUBJ
      IF FOUND()
         TXT = SUBJ + ROOM + STIME + STR(TEACHID)
         SELE 1  // แฟม TEACHER.DBF
         LOCATE FOR TEACHID = SUBJECT->TEACHID
         IF FOUND()
            TXT = TXT + TNAME + TSURN
         ENDIF
         IF GASKPRT = 2 ;  DEVPOS(PROW()+1,1)
            ELSE ; DEVPOS(ROW()+1,1) ; ENDIF
         DEVOUT(TXT) // น ําขอความทั้งหมดมารวมกันกอนพิมพ
      ENDIF
      SELE 3
      SKIP
   ENDDO
   IF GASKPRT = 2 ; EJECT ; ENDIF
   SET DEVICE TO SCREEN

   INKEY(0)
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.27
&   ผลลัพธ
// พมิพรายงานผลการเรียนตลอดหลักสูตร โดยระบุรหัสนักศึกษา
// รายงานแสดงชื่อวิชา และเกรดที่นักศึกษาไดรับ
// บรรทัดสุดทั้งแสดงการนับเกรดที่นักศึกษาได A และ F
// แสดงใหเห็นวา สามารถใช SET FILTER แทนค ําสั่ง LOCATE ได
// และการพิมพขอมูลทางเครื่องพิมพดวย SET PRINT ON และ ?
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE R0303
   DO SORTREG1   // จดัเรียงตามป ภาค และรหัส
   SELE 1 ; USE IDSTD
   SELE 2 ; USE REGIST
      _IDSTD = 0
      @ 0,5 SAY "รหสันกัศึกษาที่ตองการรายงาน : " GET _IDSTD  ; READ
      GASKPRT = ASKPRT()
      IF  GASKPRT = 2 ; SET PRINT ON ;  ENDIF
      CLS ; _RYEAR = 0  ;  _RSEM = '0'
      SET FILTER TO IDSTD = _IDSTD
      COUNT TO CNTSUBJ  // น ําไปพิมพเปนผลลัพธตอนทายโปรแกรม
      COUNT TO CNTA   FOR RGRADE = 4  // นับเฉพาะเกรด A
      COUNT TO CNTF   FOR RGRADE = 0  // นับเฉพาะเกรด F
      GO TOP  // ค ําสัง่ COUNT ท ําใหตัวชี้เรคอรดเปลี่ยน จึงตองเซตตัวชี้ที่นี่
   SELE 1
      SET FILTER TO IDSTD = _IDSTD  ;    GO TOP
      ? IDSTD,INAME,ISURN,IMAJOR,ISTAT
   SELE 2
   DO WHILE !EOF() .AND. CNTSUBJ > 0
      IF RYEAR <> _RYEAR .OR. RSEM <> _RSEM
         ? RSEM , "/" , STR(RYEAR)
         _RYEAR = RYEAR
         _RSEM = RSEM

      ENDIF
      _GRADE = GRADECHR(RGRADE)
      ? " | " , SUBJ , " | " , _GRADE , " | "
      SKIP
   ENDDO
   ? "จ ํานวนวิชาที่นักศึกษาลงทะเบียน : " , STR(CNTSUBJ)
   ? "จ ํานวนวิชาที่ไดเกรด A : " , STR(CNTA)
   ? "จ ํานวนวิชาที่ไดเกรด F : " , STR(CNTF)
   IF GASKPRT = 2 ; EJECT ; ENDIF
   SET PRINT OFF
   INKEY(0)
RETURN
: ตัวอยางที่ 5.28
&   ผลลัพธ
// ปรบัปรงุแฟมอาจารย อยางงาย ๆ ดวยฟงกชัน BROWSE
// โดยเลือกกลุมขอมูลตามฟลดที่ตองการแกไขได
PROCEDURE R0304
   @ 1,5 SAY "เลอืกกลุมขอมูลที่ทานตองการปรับปรุง ตามฟลดใด ?"
   @ 2,5 PROMPT "1. รหัส"
   @ 3,5 PROMPT "2. ชื่อ"
   @ 4,5 PROMPT "3. สกุล"
   @ 5,5 PROMPT "4. อายุ"
   @ 6,5 PROMPT "5. เพศ"
   MENU TO FTEACH
   IF FTEACH > 0
      _KEYIN = "?                     "
      @ FTEACH + 1,20 SAY ":>" GET _KEYIN ; READ // รบัคาที่ตองการคนหา
      IF FTEACH = 4 .OR. FTEACH = 1
         _KEYCHK = VAL(_KEYIN)
      ELSE
         _KEYCHK = RTRIM(_KEYIN)
      ENDIF
   ENDIF
   USE TEACHER

   IF FTEACH = 1 ; SET FILTER TO TEACHID = _KEYCHK ; ENDIF
   IF FTEACH = 2 ; SET FILTER TO TNAME = _KEYCHK ; ENDIF
   IF FTEACH = 3 ; SET FILTER TO TSURN = _KEYCHK ; ENDIF
   IF FTEACH = 4 ; SET FILTER TO TAGE = _KEYCHK ; ENDIF
   IF FTEACH = 5 ; SET FILTER TO TSEX = _KEYCHK ; ENDIF
   COUNT TO CNTREC
   IF CNTREC > 0
      GO TOP
      BROWSE(7,5,21,75)
      PACK
   ELSE
      @ 7,5 SAY "ไมพบเรคอรด หรือกลุมเรคอรดที่ทานตองการปรับปรุง"
      INKEY(0)
   ENDIF
RETURN

5.4.5 โปรแกรมตาง ๆ ใน SUB04.PRG
โปรแกรมใน SUB04.PRG นี้มีโปรแกรมยอยทั้งหมด 5 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบงาน

ประจ ําภาคเรียน เชน โปรแกรมพิมพรายชื่อนักศึกษาติดหนาหองสอบในการสอบกลางภาค และปลายภาค โปรแกรม
พมิพรายชือ่นกัศกึษาแยกตามสถานภาพ หรือแยกตามสาขา  โปรแกรมพิมพใบรายงานผลการเรียนประจํ าภาคเรียน
โปรแกรมปรับปรุงเกรดนักศึกษาเมื่ออาจารยสงเกรด เปนตน

ระบบงานประจ ําภาคเรียน เปนระบบที่มีงานที่ตองทํ าในทุกภาคเรียน และสวนใหญเปนงานประจํ า เชน การ
จดัท ํารายงานแยกตามสถานภาพ การพิมพรายชื่อติดหนาหองสอบ หรือการปอนเกรดนักศึกษาเมื่อสิ้นภาคเรียน บาง
ครัง้ผูออกแบบระบบอาจน ําโปรแกรมในระบบอื่น ๆ มารวมในระบบนี้ได  เพราะระบบนี้เปนระบบใหญมีงานมาก แตผู
ออกแบบอาจแยกงานบางอยางไปสรางระบบยอยใหม  เพื่อท ําใหลดความซับซอนได

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB04.PRG ทั้งหมด 5 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.29 ถึง ตัวอยางที่ 5.33 ( R0401 - R0405 )
: ตัวอยางที่ 5.29
&   ผลลัพธ
// พมิพรายชื่อนักศึกษาแตละวิชาติดหนาหองสอบ
// รายงานแสดงล ําดับ รหัสวิชา รหัสนักศึกษา ชื่อ สกุล และเบอรเกาอี้ในหองสอบ
// โดยสมมติใหหองสอบกับหองเรียนใชหองเดียวกัน
PROCEDURE R0401
   DO SORTREG2 // ใชจดัเรียงขอมูลตามวิชา ป ภาค และรหัส อยูใน MAIN.PRG

   SELE 1 ; USE REGIST
      SET FILTER TO RYEAR = 1996 .AND. RSEM = '1'
      GO TOP
   SELE 2 ; USE IDSTD
   SELE 3 ; USE ROOM
   SELE 4 ; USE SUBJECT
   SELE 1
   _SUBJOLD = SPACE(8)
   IIDSTD := ISUBJ := 0
   DO WHILE !EOF()
      _IDSTD = IDSTD
      IIDSTD++
      IF SUBJ <> _SUBJOLD // ใชพิมพหัวของหนาแตละหนา
         ISUBJ++    // ใชนับวิชา
         IIDSTD = 1 // ใชนับนักศึกษาในวิชา
         INKEY(0.5)
         ? '*',ISUBJ,SUBJ
         _SUBJOLD = SUBJ
         SELE 4 // แฟมวิชา ใชคนหาหองเรียน
            LOCATE FOR SUBJ = REGIST->SUBJ
            IF FOUND()
               ?? ROOM  // หองสอบของวิชานั้น
            ENDIF
      ENDIF
      SELE 2 // แฟมนักศึกษา ใชคนหาชื่อ และสาขา
         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            ? IIDSTD,IDSTD,INAME,ISURN
         ELSE
            ? IIDSTD,_IDSTD  // เพราะรหัสนี้ไมมีใน IDSTD จึงตองใชตัวแปรแทน
         ENDIF
      SELE 3 // แฟมหองเรียน ใชคนหาเลขที่โตะ
         LOCATE FOR SUBJECT->ROOM = ROOM .AND. RCSEQ = IIDSTD

         IF FOUND()
            ?? "CHAIR : ",RCHAIR
         ENDIF
      SELE 1 // แฟมลงทะเบียน เปนแฟมหลักของรายงานนี้
      SKIP
   ENDDO
   INKEY(0)
   CLOSE ALL // ตอง CLOSE ALL เพราะเปดแฟมไวหลายพื้นที่
RETURN
: ตัวอยางที่ 5.30
&   ผลลัพธ
// พมิพรายชื่อแยกตามสถานภาพ โดยครั้งแรกแสดงผลทางจอภาพทันที
// ใชค ําสั่ง SET PRINT ON และ ? ในการจัดพิมพ
// ถากด P จะพิมพขอมูลที่พิมพไปแลวทางเครื่องพิมพทันที
// หากกด P อีกครั้งจะพิมพออกทางเครื่องพิมพอีก จนกวาจะกดปุมอื่น
PROCEDURE R0402
   DO SORTID2   // ใชจัดเรียงตามสถานภาพ และรหัส
   _ISTATO = " "
   USE IDSTD
   CHKPRT = "P"
   YESPRINTON = "N"
   DO WHILE CHKPRT = "P"
      WHILE !EOF()
         IF ISTAT <> _ISTATO
            ? "======================"
            ? "CURRENT STATUS : ",ISTAT
            _ISTATO = ISTAT
         ENDIF
         ? IDSTD,INAME,ISURN,IMAJOR
         SKIP
      END
      WAIT "ถาอยากพิมพใหกด P" // บรรทัดนี้ใหผลลัพธทางเครื่องพิมพ และจอภาพ
      CHKPRT = UPPER(CHR(LASTKEY()))
      IF CHKPRT = "P" // ตรวจสอบความตองการรายงานทางเครื่องพิมพ

         YESPRINTON = "Y"
         SET PRINT ON
         GO TOP ; CLS
      ENDIF
   ENDDO
   IF  YESPRINTON = "Y" ; EJECT ; ENDIF
   SET PRINT OFF
RETURN
: ตัวอยางที่ 5.31
&   ผลลัพธ
// พมิพรายชื่อแยกตามสาขา และจัดเรียงตามชื่อ
// จดักลุมขอมูลตามสาขา โดยรายงานแสดง รหัสนักศึกษา ชื่อ สกุล และสถานภาพ
// ผลลพัธแสดงทางเครื่องพิมพดวยการกด P เมื่อแสดงขอมูลทางจอภาพไปแลว
// จ ํานวนบรรทัดตอหนาของจอภาพระบุไวเปน 10 และเครื่องพิมพเปน 30
PROCEDURE R0403
   DO SORTID3  // ใชจัดเรียงตามสาขา และชื่อ
   _IMAJORO = 0
   USE IDSTD
   CHKPRT = "P"
   YESPRINTON = "N"
   LN = 1 ; LNSCR = 10 ; LNPRT = 30
   @ LN,1 SAY "ขอมลูชดุนี้เรียงตามจัดกลุมตามสาขา และเรียงตามชื่อ"
   DO WHILE CHKPRT = "P"
      WHILE !EOF()
         IF IMAJOR <> _IMAJORO
            LN++ ; @ LN,1 SAY "======================"
            LN++ ; @ LN,1 SAY "CURRENT MAJOR : " + STR(IMAJOR)
            _IMAJORO = IMAJOR
         ENDIF
         LN++ ; @ LN,1 SAY STR(IDSTD) + INAME + ISURN + ISTAT
         IF (YESPRINTON = "N" .AND. LN >= LNSCR) .OR. ;
            (YESPRINTON = "Y" .AND. LN >= LNPRT)
            INKEY(1)
            CLS // เงือ่นไขนี ้ท ําใหจํ านวนบรรทัดตอจอภาพ และเครื่องพิมพตางกัน

            LN = 0
         ENDIF
         SKIP
      END
      IF  YESPRINTON = "Y" ; EJECT ; ENDIF
      // ค ําสั่ง SET DEVICE TO PRINTER ไมสงผลตอค ําสั่ง WAIT
      WAIT "ถาอยากพิมพใหกด P" // สอบถามการพิมพออกเครื่องพิมพ
      CHKPRT = UPPER(CHR(LASTKEY()))
      IF CHKPRT = "P"
         YESPRINTON = "Y"
         SET DEVICE  TO PRINTER
         GO TOP ; CLS
      ENDIF
   ENDDO
   SET DEVICE TO SCREEN
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.32
&   ผลลัพธ
// ปรบัปรงุเกรดนักศึกษาเมื่ออาจารยสงเกรด โดยระบุปการศึกษา ภาค และวิชา
// รอรบัเกรดใหม แลวปรับปรุงทีละคน จนหมดขอมูลที่ตรงกับเงื่อนไข
PROCEDURE R0404
   DO SORTREG1 // ใชจัดเรียงขอมูลตามป ภาค รหัส และวิชา
   _RYEAR = 1996
   _RSEM = '1'
   _SUBJ = SPACE(8)
   USE REGIST
   DO WHILE !FOUND()
      @ 5,5 SAY "ปรบัปรุงเกรดนักศึกษา"
      @ 6,5 SAY "ปการศึกษา " GET _RYEAR PICT "9999"
      @ 7,5 SAY "ภาคการศึกษา " GET _RSEM
      @ 8,5 SAY "วชิาที่ตองการปอนเกรด " GET _SUBJ PICT "@!"
      READ
      LOCATE FOR _RYEAR = RYEAR .AND. _RSEM = RSEM .AND. SUBJ = _SUBJ

   ENDDO
   SET FILTER TO RYEAR = _RYEAR .AND. RSEM = _RSEM .AND. SUBJ = _SUBJ
   COUNT TO CNTALL
   GO TOP
   CNT = 1
   @ 9,5 SAY "จ ํานวนนกัศึกษาทั้งหมดในวิชานี้ : "+STR(CNTALL)
   DO WHILE !EOF() .AND. LASTKEY() != 27
      _GRADE = "  "
      WHILE _GRADE = "  "
         @ 10,5 SAY STR(CNT) +": "+ STR(IDSTD) +" : "+ STR(RGRADE) +": ";
            GET _GRADE
         READ
      END
      _RGRADE = GRADENUM(_GRADE) // แปลงเกรดตัวอักษรเปนตัวเลข
      REPLACE RGRADE  WITH _RGRADE
      SKIP
      CNT++
   ENDDO
   CLOSE ALL
RETURN
: ตัวอยางที่ 5.33
&   ผลลัพธ
// พมิพใบรายงานผลการเรียนประจ ําภาค โดยระบุปการศึกษา และภาคการศึกษา
// จดักลุมตามรหัสนักศึกษา เพื่อพิมพใบรายงานผลการเรียนใหนักศึกษาแตละคน
// รายงานแสดงล ําดับ รหัสวิชา และเกรด
// บรรทดัสดุทายแสดงเกรดเฉลี่ยประจ ําภาคเรียนของนักศึกษาคนนั้น
PROCEDURE R0405
   _RYEAR = 1996
   _RSEM = '1'
   @ 5,5 SAY "ปการศึกษา" GET _RYEAR PICT "9999"
   @ 6,5 SAY "ภาคการศึกษา" GET _RSEM
   READ
   USE REGIST
   SET FILTER TO RYEAR = _RYEAR .AND. RSEM = _RSEM

   GO TOP
   _GPA = 0
   _CNTSUBJ = 0
   _SUMPOINT = 0
   _IDSTDO = 0
   CNTID = 0
   CNT = 1
   DO WHILE !EOF() .AND. LASTKEY() != 27
      IF IDSTD <> _IDSTDO
         IF CNT <> 1
            _NEXTIDSTD = IDSTD
            COUNT TO _CNTSUBJ FOR RYEAR = _RYEAR .AND. RSEM = _RSEM .AND.;
              IDSTD = _IDSTDO .AND. RGRADE <= 4
            SUM RGRADE TO _SUMPOINT FOR RYEAR = _RYEAR .AND. RSEM = _RSEM ;
              .AND. IDSTD = _IDSTDO .AND. RGRADE <= 4
            _GPA  = _SUMPOINT / _CNTSUBJ
            @ ROW()+1,5 SAY "เกรดเฉลี่ยประจ ําภาคเรียนนี้คือ "+STR(_GPA)
            INKEY(3)
            LOCATE FOR IDSTD = _NEXTIDSTD // เพราะ COUNT จะเปลี่ยนตัวชี้ใหม
         ENDIF  // จงึตองใช LOCATE เลื่อนตัวชี้ใหถูกตองอีกครั้ง
         CNT = 1
         CNTID++
         @ 7,5 CLEAR TO 22,75
         SETPOS(7,1)
         @ ROW()+1,5 SAY STR(CNTID) + " : " + STR(IDSTD)
         @ ROW()+1,5 SAY "================================"
         _IDSTDO = IDSTD
      ENDIF
      _GRADE = GRADECHR(RGRADE)  //  แปลงเกรดตัวเลขเปนตัวอักษร
      @ ROW()+1,10 SAY STR(CNT) + " : " + SUBJ + " : " + _GRADE
      SKIP
      CNT++
   ENDDO

   CLOSE ALL
RETURN

5.4.6 โปรแกรมตาง ๆ ใน SUB05.PRG
โปรแกรมใน SUB05.PRG นี้มีโปรแกรมยอยทั้งหมด 5 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบปรับปรุง

และพมิพขอมูลจากแฟม เชน โปรแกรมปรับปรุงแฟมและพิมพ ของแฟมในแบบตาง ๆ  ท ําใหสามารถเลือกเปดแฟมใน
ระบบ โดยระบบประกอบดวย แฟมนักศึกษา แฟมวิชา แฟมผูสอน แฟมลงทะเบียน และแฟมหอง

ระบบงานปรบัปรุงและพิมพจากแฟมอยางงาย ๆ ชวยทํ าใหการปรับปรุงขอมูลทํ าไดทันท ีแตไมเหมาะกับการ
น ําไปใชงานจรงิ เพราะการปรับปรุงขอมูลจะทํ าใหควบคุมความปลอดภัยหรือความถูกตองทํ าไดยาก ดังนั้นการปรับ
ปรงุจงึควรผานโปรแกรมที่อนุญาติใหแกไขขอมูล โดยมีการตรวจสอบความถูกตองของขอมูลอยางดี กอนจะจัดเก็บขอ
มลูลงแฟม  และการพิมพขอมูลควรที่จะเชื่อมขอมูลกับแฟมตาง ๆ ท ําใหเกิดรายงานทีอ่านเขาใจไดโดยงายเพราะการ
น ําแฟมเพยีงแฟมเดียวมาพิมพจะไมสมบูรณนัก เพราะมีแฟมหลาย ๆ แฟมที่เก็บขอมูลเปนรหัส  ท ําใหตองนํ ารหัสนั้น
ไปเชือ่มกับแฟมอื่น จึงจะไดขอมูลที่อานแลวเขาใจไดทันที

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SUB05.PRG ทั้งหมด 5 โปรแกรม โดยเริ่มจากตัว
อยางที ่5.34 ถึง ตัวอยางที่ 5.38 ( R0501 - R0505 )
: ตัวอยางที่ 5.34
&   ผลลัพธ
// การปรับปรุงแฟมนักศึกษา
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0501
   _IDSTD = 0
   X = ALERT("ปรบัปรงุแฟมนักศึกษา",{'พิมพขอมูล','เพิ่ม','ลบ','แกไข'})
   DO CASE
      CASE X = 0 ; WAIT "ไมไดเลือกตัวเลือกใดเลย"
      CASE X = 1
      // พมิพขอมูลจากแฟมนักศึกษา
         @ 10,5 SAY "พมิพขอมูลจากแฟมนักศึกษา"
         GASKPRT = ASKPRT()
         IF GASKPRT = 0 ; RETURN ; ENDIF
         USE IDSTD
         IF GASKPRT = 1
            CLS
            LIST IDSTD,INAME,ISURN,IMAJOR,ISTAT
            INKEY(0)

         ENDIF
         IF GASKPRT = 2        // ผลออกทั้งจอภาพและเครื่องพิมพ
            LIST IDSTD,INAME,ISURN,IMAJOR,ISTAT TO PRINTER
            EJECT
         ENDIF
      CASE X = 2
      // เพิ่มขอมูลในแฟมนักศึกษา
         USE IDSTD
         APPEND BLANK
         @ 10,5 SAY "เพิม่ขอมูลในแฟมนักศึกษาใหม"
         @ 11,5 SAY "รหัส" GET IDSTD
         @ 12,5 SAY "ชือ่" GET INAME
         @ 13,5 SAY "สกุล" GET ISURN
         @ 14,5 SAY "สาขา" GET IMAJOR
         @ 15,5 SAY "สถานภาพ" GET ISTAT
         READ
         REPLACE IDSTD WITH IDSTD ,INAME WITH INAME ,ISURN WITH ISURN,;
                 IMAJOR WITH IMAJOR,ISTAT WITH ISTAT
      CASE X = 3
      // ลบขอมูลจากแฟมนักศึกษา
         @ 10,5 SAY "ลบขอมูลจากแฟมนักศึกษา"
         USE IDSTD
         @ 11,5 SAY "รหสันักศึกษาที่ตองการลบ : " GET _IDSTD ; READ
         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            DELETE
            PACK
         ENDIF
      CASE X = 4
      // แกไขขอมูลจากแฟมนักศึกษา
         @ 10,5 SAY "แกไขขอมูลจากแฟมนักศึกษา"
         USE IDSTD
         @ 11,5 SAY "รหสันักศึกษาที่ตองการแกไข : " GET _IDSTD ; READ

         LOCATE FOR IDSTD = _IDSTD
         IF FOUND()
            @ 12,5 SAY "ชื่อ" GET INAME
            @ 13,5 SAY "สกุล" GET ISURN
            @ 14,5 SAY "สาขา" GET IMAJOR
            @ 15,5 SAY "สถานภาพ" GET ISTAT
            READ
            REPLACE IDSTD WITH _IDSTD ,INAME WITH INAME ,ISURN WITH ISURN,;
                 IMAJOR WITH IMAJOR,ISTAT WITH ISTAT
         ENDIF
   ENDCASE
RETURN
: ตัวอยางที่ 5.35
&   ผลลัพธ
// การปรับปรุงแฟมวิชา
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0502
   _SUBJ = SPACE(8)
   _TEACHID = 0
   _ROOM = SPACE(4)
   _STIME = SPACE(11)
   _SREGIST := _SLIMIT := 0
   X = ALERT("ปรบัปรุงแฟมวิชา",{'พิมพขอมูล','เพิ่ม','ลบ','แกไข'})
   DO CASE
      CASE X = 0 ; WAIT "ไมไดเลือกตัวเลือกใดเลย"
      CASE X = 1
      // พมิพขอมูลจากแฟมวิชา
         @ 4,5 SAY "พมิพขอมูลจากแฟมวิชา"
         GASKPRT = ASKPRT()
         IF GASKPRT = 0 ; RETURN ; ENDIF
         USE SUBJECT
         IF GASKPRT = 1
            AR = {"SUBJ","TEACHID","ROOM","STIME","SREGIST","SLIMIT"}
            DBEDIT(5,5,20,75,AR)

         ENDIF
         IF GASKPRT = 2        // ผลออกทั้งจอภาพและเครื่องพิมพ
            LIST SUBJ,TEACHID,ROOM,STIME,SREGIST,SLIMIT
            EJECT
         ENDIF
      CASE X = 2
      // เพิ่มขอมูลในแฟมวิชา
         USE SUBJECT
         @ 10,5 SAY "เพิม่ขอมูลในแฟมวิชาใหม"
         @ 11,5 SAY "รหัสวิชา" GET _SUBJ
         @ 12,5 SAY "รหัสอาจารยผูสอน" GET _TEACHID
         @ 13,5 SAY "หองเรียน" GET _ROOM
         @ 14,5 SAY "เวลาเรียน" GET _STIME
         @ 15,5 SAY "จ ํานวนที่ลงทะเบียน" GET _SREGIST
         @ 16,5 SAY "จ ํานวนสูงสุดที่อนุญาต"ิ GET _SLIMIT
         READ
         IF ASKUPD() = 1
            APPEND BLANK
            REPLACE SUBJ WITH _SUBJ,TEACHID WITH _TEACHID,ROOM WITH _ROOM,;
                    STIME WITH _STIME,SREGIST WITH _SREGIST,SLIMIT WITH _SLIMIT
         ENDIF
      CASE X = 3
      // ลบขอมูลจากแฟมวิชา
         @ 10,5 SAY "ลบขอมูลจากแฟมวิชา"
         USE SUBJECT
         @ 11,5 SAY "รหสัวิชาที่ตองการลบ : " GET _SUBJ ; READ
         LOCATE FOR SUBJ = _SUBJ
         IF FOUND()
            @ 12,5 SAY SUBJ
            @ 13,5 SAY TEACHID
            @ 14,5 SAY ROOM
            @ 15,5 SAY STIME
            @ 16,5 SAY SREGIST

            @ 17,5 SAY SLIMIT
            IF ASKUPD() = 1
               DELETE ; PACK
            ENDIF
         ENDIF
      CASE X = 4
      // แกไขขอมูลจากแฟมวิชา
         @ 10,5 SAY "แกไขขอมูลจากแฟมวิชา"
         USE SUBJECT
         @ 11,5 SAY "รหัสวิชา : " GET _SUBJ ; READ
         LOCATE FOR SUBJ = _SUBJ
         IF FOUND()
            _SUBJ = SUBJ
            _TEACHID = TEACHID
            _ROOM = ROOM
            _STIME = STIME
            _SREGIST = SREGIST
            _SLIMIT = SLIMIT
            @ 11,5 SAY "รหัสวิชา : " GET _SUBJ
            @ 12,5 SAY "รหัสอาจารยผูสอน" GET _TEACHID
            @ 13,5 SAY "หองเรียน" GET _ROOM
            @ 14,5 SAY "เวลาเรียน" GET _STIME
            @ 15,5 SAY "จ ํานวนที่ลงทะเบียน" GET _SREGIST
            @ 16,5 SAY "จ ํานวนสูงสุดที่อนุญาต"ิ GET _SLIMIT
            READ
            IF ASKUPD() = 1
               REPLACE SUBJ WITH _SUBJ,TEACHID WITH _TEACHID,;
                       ROOM WITH _ROOM,STIME WITH _STIME,;
                       SREGIST WITH _SREGIST,SLIMIT WITH _SLIMIT
            ENDIF
         ENDIF
   ENDCASE
RETURN

: ตัวอยางที่ 5.36
&   ผลลัพธ
// การปรับปรุงแฟมผูสอน
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0503
   USE TEACHER
   BROWSE(5,5,22,75)
   PACK
RETURN
: ตัวอยางที่ 5.37
&   ผลลัพธ
// การปรับปรุงแฟมลงทะเบียน
// โดยเลือกจัดเรียงขอมูลดวยค ําสั่ง SORT
// ปรบัปรุงขอมูลโดยใชคํ าสั่ง BROWSE
PROCEDURE R0504
   @ 4,5 SAY "การปรับปรุงแฟมลงทะเบียน"
   @ 9,5 SAY "สัง่จัดเรียงกอนปรับปรุงขอมูล"
   @ 10,5 PROMPT "1. จดัเรยีงตามป ภาค รหัสนักศึกษา และวิชา"
   @ 11,5 PROMPT "2. จดัเรยีงตามวิชา ป ภาค และรหัสนักศึกษา"
   @ 12,5 PROMPT "3. จดัเรยีงตามรหัสนักศึกษา ป ภาค และวิชา"
   @ 13,5 PROMPT "4. จดัเรยีงตามเกรด ป ภาค วิชา และรหัสนักศึกษา"
   @ 14,5 PROMPT "5. จดัเรยีงตามรหัสนักศึกษา และวิชา จากมากไปนอย"
   @ 15,5 PROMPT "6. ไมจัดเรียงใหม"
   MENU TO _SORTORDER
   IF _SORTORDER > 0 .AND. _SORTORDER < 6
      USE REGIST
      DO CASE
         CASE _SORTORDER = 1
            SORT ON RYEAR,RSEM,IDSTD,SUBJ TO TEMPF
         CASE _SORTORDER = 2
            SORT ON SUBJ,RYEAR,RSEM,IDSTD TO TEMPF
         CASE _SORTORDER = 3
            SORT ON IDSTD,RYEAR,RSEM,SUBJ TO TEMPF
         CASE _SORTORDER = 4
            SORT ON GRADE,RYEAR,RSEM,SUBJ,IDSTD TO TEMPF

         CASE _SORTORDER = 5
            SORT ON IDSTD/D ,SUBJ TO TEMPF
       ENDCASE
       USE TEMPF
       COPY TO REGIST
       CLOSE
   ENDIF
   USE REGIST
   BROWSE(5,5,22,75)
   PACK
RETURN
: ตัวอยางที่ 5.38
&   ผลลัพธ
// การปรับปรุงแฟมหอง
// โดยเลือกพิมพ เพิ่ม ลบ หรือแกไขได
PROCEDURE R0505
   _ROOM = SPACE(4)
   _RCSEQ = 0
   _RCHAIR = SPACE(2)
   X = ALERT("ปรบัปรุงแฟมหอง",{'พิมพขอมูล','เพิ่ม','ลบ','แกไข'})
   DO CASE
      CASE X = 0 ; WAIT "ไมไดเลือกตัวเลือกใดเลย"
      CASE X = 1
      // พมิพขอมูลจากแฟมหอง
         @ 4,5 SAY "พมิพขอมูลจากแฟมหอง"
         GASKPRT = ASKPRT()
         IF GASKPRT = 0 ; RETURN ; ENDIF
         USE ROOM
         IF GASKPRT = 1
            DISPLAY ROOM,RCSEQ,RCHAIR ALL
            INKEY(5)
         ENDIF
         IF GASKPRT = 2        // ผลออกทั้งจอภาพและเครื่องพิมพ
            FOR I = 1 TO  RECCOUNT()

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                            บทที่ 5  :  กรณศีกึษาระบบงานทะเบียน                                                      281

               DISPLAY I,ROOM,RCSEQ,RCHAIR TO PRINTER
               SKIP
            NEXT
            EJECT
         ENDIF
      CASE X = 2
      // เพิ่มขอมูลในแฟมหอง
         USE ROOM
         @ 10,5 SAY "เพิม่ขอมูลในแฟมหองใหม"
         SET FORMAT TO R0505FORM
         READ
         SET FORMAT TO
         IF ASKUPD() = 1
            APPEND BLANK
            REPLACE ROOM WITH _ROOM,RCSEQ WITH _RCSEQ,RCHAIR WITH _RCHAIR
         ENDIF
      CASE X = 3
      // ลบขอมูลจากแฟมหอง
         @ 9,5 SAY "ลบขอมูลจากแฟมหอง"
         USE ROOM
         @ 10,5 SAY "รหสัหองที่ตองการลบ : " GET _ROOM
         @ 11,5 SAY "ล ําดับเกาอี้ : " GET _RCSEQ
         READ
         LOCATE FOR ROOM = _ROOM .AND. RCSEQ = _RCSEQ
         IF FOUND()
            @ 12,5 SAY RCHAIR
            IF ASKUPD() = 1
               DELETE ; PACK
            ENDIF
         ENDIF
      CASE X = 4
      // แกไขขอมูลจากแฟมหอง
         @ 7,5 SAY "แกไขขอมูลจากแฟมหอง"

         USE ROOM
         @ 8,5 SAY "รหสัหองที่ตองการแกไข : " GET _ROOM
         @ 9,5 SAY "ล ําดับเกาอี้ : " GET _RCSEQ
         READ
         LOCATE FOR ROOM = _ROOM .AND. RCSEQ = _RCSEQ
         IF FOUND()
            _ROOM = ROOM
            _RCSEQ = RCSEQ
            _RCHAIR = RCHAIR
            SET FORMAT TO R0505FORM
            READ
            SET FORMAT TO    // ตองยกเลิก FORMAT เพราะปองกันการ READ แบบอื่น
            IF ASKUPD() = 1
               REPLACE ROOM WITH _ROOM,RCSEQ WITH _RCSEQ,RCHAIR WITH _RCHAIR
            ENDIF
         ENDIF
   ENDCASE
   PROCEDURE R0505FORM
   @ 10,5 SAY "หองเรียน" GET _ROOM
   @ 11,5 SAY "ล ําดับของเกาอี้" GET _RCSEQ
   @ 12,5 SAY "เลขที่เกาอี"้ GET _RCHAIR
   RETURN
RETURN

